
А.П. Немытых

Лекции по языку программирования Рефал

Мы представляем конспекты лекций по функциональному языку про-
граммирования Рефал. Лекции были прочитаны автором в 2006 году в г.
Переславле-Залесском. Библ. 10 наим.

Содержание

Содержание . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 118

1. Данные языка программирования Рефал

(Лекция №1) . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 120

1.1. Уравнения в свободной полугруппе . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 120

1.2. Системы уравнений в свободной полугруппе . . . . . . . . . . . . . . . . . . . . 121

1.2.1. Леса корневых деревьев . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 122

1.3. Домашнее задание . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 123

2. Функции индуктивные по Кушниренко

(Лекция №2) . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 124

2.1. Функции индуктивные по Кушниренко . . . . . . . . . . . . . . . . . . . . . . . . . . 124

2.2. Примеры программ на языке Рефал . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 125

2.2.1. Как выполнить программу на языке Рефал . . . . . . . . . . . . . . . . 127

2.3. Домашнее задание . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 128

3. Унарный конструктор построения дерева. Рефал-машина.

(Лекция №3) . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 128

3.1. Парные скобки как унарный конструктор построения дерева . . . 129

3.2. Абстрактная Рефал-машина . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 130

3.3. Домашнее задание . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 132

4. Арифметика

(Лекция №4) . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 133

4.1. Представление целых чисел . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 133

4.2. Встроенные функции арифметики . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 133

4.3. Два типа рекурсии . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 134

4.4. Фибоначчиева система счисления . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 134

4.5. Домашнее задание . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 136

5. Функциональность. Ввод-вывод.

(Лекция №5) . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 136

5.1. Встроенный ввод-вывод . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 137

5.2. Пример А.В. Корлюкова. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 138

5.3. Домашнее задание . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 139

c○ А.П. Немытых, 2014



ЛЕКЦИИ ПО ЯЗЫКУ ПРОГРАММИРОВАНИЯ РЕФАЛ 119

6. Алгоритмическая полнота. Алгорифмы Маркова.

(Лекция №6) . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 146

6.1. Алгоритмическая полнота . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 146

6.2. Алгорифмы Маркова . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 147

6.3. Домашнее задание . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 149

7. Вызов функции по определению

(Лекция №7) . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 150

7.1. Блоки . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 150

7.2. Домашнее задание . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 153

8. Рефал как метаязык программирования

(Лекция №8) . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 153

8.1. Самоприменимые программы . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 155

8.2. Функция применения . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 156

8.3. Домашнее задание . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 156

9. Рекурсивные условия

(Лекция №9) . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 156

9.1. Другая интерпретация знака запятой . . . . . . . . . . . . . . . . . . . . . . . . . . . . 157

9.2. Рекурсивные условия выбора предложения . . . . . . . . . . . . . . . . . . . . . . 157

9.2.1. Откаты при выборе предложения . . . . . . . . . . . . . . . . . . . . . . . . . . . 158

9.2.2. Локальное присваивание . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 161

9.3. Домашнее задание . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 161

10. Сложность Колмогорова

(Лекция №10) . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 162

10.1. Сложность текста . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 162

10.2. Структурные скобки и кодирование разделения аргументов . . . 163

10.3. Теорема Колмогорова . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 163

10.4. Домашнее задание . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 164

Список литературы . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 165



120 А.П. НЕМЫТЫХ

§ 1. Данные языка

программирования Рефал

(Лекция №1)

РЕкурсивный Функциональный АЛгоритмический язык РЕФАЛ1 был раз-

работан В.Ф. Турчиным как универсальный метаязык программирования в

60-х годах. Первая реализация появилась в 1968 г. Рефал – строгий язык пер-

вого порядка, основанный на алгорифмах Маркова, имеет два неизоморфных

конструктора; один из которых – приписывание (конкатенация) – ассоциати-

вен, а другой – унарный – используется для построения структуры произволь-

ного дерева. Ассоциативность приписывания превращает множество данных

Рефала в моноид. Структура этого моноида отражается в синтаксисе про-

грамм и операционной семантике Рефала.

1.1. Уравнения в свободной полугруппе. Моноидом называется полу-

группа с единицей.

Пусть 𝑀 – непустое множество. Тогда свободным моноидом, порождённым

𝑀 , называется множество всех конечных последовательностей (включая пу-

стую последовательность) элементов из 𝑀 с полугрупповой операцией, зада-

ваемой приписыванием последовательностей, т. е.

{𝑎1, . . . , 𝑎𝑛} ∘ {𝑏1, . . . , 𝑏𝑚} = {𝑎1, . . . , 𝑎𝑛, 𝑏1, . . . , 𝑏𝑚}

Далее пустую последовательность будем обозначать ничем (отсутствием како-

го либо знака).

Рассмотрим свободный моноид, порождённый множеством букв Кирилли-

цы и латинского алфавита. Назовём этот моноид алфавитной полугруппой.

Элементы алфавитной полугруппы будем писать в одинарных кавычках. На-

пример, ’ы’. Слово, заключённое в одинарные кавычки, будем считать сокра-

щением записи последовательности букв, входящих в это слово. Например,

’полугруппа’ = ’п’ ’о’ ’л’ ’у’ ’г’ ’р’ ’у’ ’п’ ’п’ ’а’

Задача 1. Решить уравнения в алфавитной полугруппе (где 𝑥, 𝑦 и 𝑧 – пе-

ременные):

а). 𝑥 ’п’ 𝑧 = ’полугруппа’;

б). 𝑥2 = 𝑥𝑥 = ’abababab’;

в). 𝑥2 ’a’ 𝑦 = ’aabaabaab’;

Иногда удобно выделять последовательности, состоящие из одного элемента.

Перед именами переменных с областью допустимых значений (О.Д.З.) – мно-

жеством одноэлементных последовательностей будем приписывать приставку

«s.», перед именами переменных с О.Д.З., совпадающим со всей полугруппой,

будем приписывать приставку «e.» Например: s.x, e.y .

1Название, данное В.Ф. Турчиным, – алгоритмический язык рекурсивных функций. Мы
предпочитаем другую расшифровку этого сокращения, избегая перестановки. Не существует
общепринятой традиции, какими буквами – прописными (и сколькими) или строчными –
писать название языка. В.Ф. Турчин также использовал разные варианты синтаксиса этого
названия. Мы будем писать Рефал, делая редкие исключения.

Задача 2. Решить уравнения в алфавитной полугруппе

а). e.x s.y ’п’ e.z = ’полугруппа’;

б). s.y e.x3 s.z = ’abababab’;

в). e.u s.z2 e.x2 e.y = ’aabaabaab’;

Теперь рассмотрим уравнения с параметрами. Аналогично переменным, па-

раметры будут двух типов, соответствующих введённым типам переменных.

Чтобы отличать параметры от переменных, мы будем помечать параметры

знаком диез «#». Например: #s.p – параметр типа «s» с именем «p», #e.q –

параметр с именем «q» и О.Д.З., совпадающим со всей полугруппой.

Задача 3. Решить уравнения с параметрами в алфавитной полугруппе

а). e.x s.y ’п’ e.z = ’пол’ #s.p ’группа’;

б). e.x ’a’ = ’a’ #e.p;

в). e.y ’a’ e.x2 = ’ab’ #s.p #s.p ’abaab’;

Задача 4. Рассмотрим свободный моноид, порождённый множеством из

’0’ и ’1’. Придумать в этом моноиде уравнение с параметрами #s.p, #s.q,

#s.r

. . . = #s.p #s.q #s.r

такое, что одно из его переменных есть s.d и решение этого уравнения имеет

вид

. . ., s.d = ’0’, . . .
тогда и только тогда, когда последовательность #s.p #s.q #s.r содержит

больше нулей, чем единиц.

1.2. Системы уравнений в свободной полугруппе. Рассмотрим систе-

му из двух уравнений в алфавитной полугруппе

{︃
e.x e.y e.z = ’abbabaab’

e.y s.u = ’bab’

Скобки ( и ) не принадлежат алфавиту. Алфавитная полугруппа ассоци-

ативна, что позволяет нам опускать эти скобки, не указывая порядок испол-

нения операции приписывания в полугруппе. Таким образом, скобки ( и )

можно использовать как метасимволы – для иных целей. Воспользуемся этим

замечанием, чтобы формально записать систему уравнений более кратко:

e.x e.y e.z (e.y s.u) = ’abbabaab’ (’bab’)

Аналогично можно записать любую систему уравнений в алфавитной полу-

группе. Например, систему

⎧
⎪⎨
⎪⎩

e.y1 ’н’ e.n ’учи’ e.y2 = ’Какая’ ’ночь’ ’Мороз’ ’трескучий’

e.z1 ’н’ e.m ’учи’ e.z2 = ’На’ ’небе’ ’ни’ ’единой’ ’тучи’

e.x1 s.k e.x2 s.k e.x3 = ’Как’ ’шитый’ ’полог’ ’синий’ ’свод’

можно записать так

e.y1 ’н’ e.n ’учи’ e.y2 (e.z1 ’н’ e.m ’учи’ e.z2) (e.x1 s.k e.x2 s.k e.x3) =

’Какая’ ’ночь’ ’Мороз’ ’трескучий’

(’На’ ’небе’ ’ни’ ’единой’ ’тучи’) (’Как’ ’шитый’ ’полог’ ’синий’ ’свод’)

или так

e.y1 ’н’ e.n ’учи’ e.y2 (e.z1 ’н’ e.m ’учи’ e.z2 (e.x1 s.k e.x2 s.k e.x3))

=

’Какая’ ’ночь’ ’Мороз’ ’трескучий’

(’На’ ’небе’ ’ни’ ’единой’ ’тучи’ (’Как’ ’шитый’ ’полог’ ’синий’ ’свод’))

Задача 5. Решите данные выше уравнения.

Пусть 𝐺 – свободный моноид. Тогда древовидной полугруппой 𝐷, порождён-

ной 𝐺, назовём наибольшее множество, которое можно получить из элементов

𝐺 посредством конечного числа применений бинарной операции приписывания

и унарной операции заключения в скобки. То есть, 𝐺 ⊂ 𝐷 и

1. если 𝑑1 ∈ 𝐷 и 𝑑2 ∈ 𝐷, то 𝑑1 ∘ 𝑑2 = 𝑑1 𝑑2 ∈ 𝐷;

2. если 𝑑 ∈ 𝐷, то (𝑑) ∈ 𝐷

Задача 6. Докажите, что древовидная полугруппа является полугруппой

с единицей относительно операции приписывания.

Пусть Ω – алфавитная полугруппа, тогда правая часть последнего, приведён-

ного выше, уравнения принадлежит к древовидной полугруппе, порождённой

Ω. Выражение

() (() ())

принадлежит любой древовидной полугруппе. Более того, любая правильная

скобочная структура (выражение, состоящее только из скобок) принадлежит

любой древовидной полугруппе.

Задача 7. (числа Каталана): Сколько существует различных правильных

скобочных структур из 𝑛 пар скобок?

1.2.1. Леса корневых деревьев. Ориентированным графом будем называть

тройку Γ = {𝑉,𝐸, 𝐼}, состоящую из конечного множества вершин 𝑉 , конечного

(возможно, пустого) множества рёбер 𝐸 и отображения инцидентности

𝐼 : 𝐸 → 𝑉 × 𝑉 , сопоставляющего каждому ребру упорядоченную пару вершин

(начало и конец ребра), которые это ребро соединяет.

Последовательность ребер 𝑒1, . . . , 𝑒𝑛 ориентированного графа назовем пу-

тем, если начало каждого последующего ребра является концом предыдущего.

Корневым деревом назовем ориентированный граф Γ с выделенной верши-

ной 𝑟 (корнем), в которую не входит ни одно ребро и из которой существует

единственный путь в любую другую вершину. Кроме того, для каждой верши-

ны из Γ множество исходящих из неё рёбер упорядочено.

Задача 8. Докажите, что в корневом дереве всякая вершина, отличная от

корня, имеет ровно одно входящее в неё ребро.

Вершина называется узлом, если из неё выходит не менее одного ребра. Мно-

жество вершин, не являющихся узлами, разобьём на два непересекающихся

множества; элементы одного из них назовём листьями, элементы другого –

срезами. Каждому листу припишем некоторое имя.

Пусть дано множество имён 𝒜. Лесом корневых деревьев 𝐹𝒜 назовём мно-

жество всех конечных последовательностей (включая пустую последователь-

ность) корневых деревьев, листья которых поименованы элементами из 𝒜.
Деревья будем изображать растущими от корня вниз. Каждому выражению

с правильной скобочной структурой можно сопоставить лес корневых деревьев.

Например,

(’abc’) ’c’ (() (’d’))

⇓
����y������

����’a’

?

����’b’

HHHHHj

����’c’

����’c’ ����y�
��	

����i
@
@@R

����
?

����’d’
Рис. 1. Прямое соответствие.

Обратно, каждому лесу корневых деревьев можно сопоставить выражение с

правильной скобочной структурой.

Задача 9. Опишите обратное соответствие.

1.3. Домашнее задание.

Задача 10. Решить уравнения в алфавитной полугруппе (где 𝑥 и 𝑦 пере-

менные)

a). 𝑦 ’a’ 𝑥2 = ’abbabaab’;

б). 𝑥 ’a’ = ’a’ 𝑥.

Задача 11. Решить уравнения в алфавитной полугруппе

а). e.y s.z e.x2 = ’abbabaab’;

б). s.x ’a’ = ’a’ s.x e.y.

Задача 12. Решить уравнения с параметрами в алфавитной полугруппе

a). s.y e.x3 s.z = ’abab’ #e.p;

б). e.x2 ’a’ e.y = #e.p #e.p ’aab’.

Задача 13. Дан свободный моноид 𝐺, порождённый множеством из букв

’a’ и ’b’. В 𝐺 рассматривается уравнение с параметрами:

e.y e.x2 e.z = #s.p #s.q #s.r #e.u

Докажите, что это уравнение имеет решение такое, что длина последователь-

ности e.x больше нуля.

Задача 14. Последовательность 𝑤𝑛 определяется индуктивно по правилу:

𝑤0 = ’a’, 𝑤1 = ’ab’, 𝑤𝑛+2 = 𝑤𝑛+1 𝑤′
𝑛+1,

где 𝑤′
𝑛+1 получается из 𝑤𝑛+1 заменой всех букв ’a’ на ’b’ и всех букв ’b’ на

’a’. Например, 𝑤2 = ’abba’, 𝑤3 = ’abbabaab’. Пусть 𝑢 = 𝑤𝑘 для некоторого

𝑘. Решить уравнения: 1) e.y ’a’3 e.z = 𝑢 ; 2) e.y ’b’3 e.z = 𝑢 .

Задача 15. Установите на вашем компьютере одну из реализаций диалек-

тов языка Рефал.

§ 2. Функции индуктивные по Кушниренко

(Лекция №2)

РЕФАЛ – РЕкурсивный Функциональный АЛгоритмический, по определе-

нию, есть рекурсивный язык программирования. Это означает, что основным

выразительным инструментом программирования на Рефале является рекур-

сия (от recursion – повторение). Понятие рекурсии тесно связано с понятием

индукции по структуре данных: Рефал-программу можно часто понимать как

доказательство некоторого утверждения о данных методом математической ин-

дукции. Мы рассмотрим индукцию по данным, построенным с помощью кон-

структора приписывания.

2.1. Функции индуктивные по Кушниренко. Скажем, что функция

𝐹 : 𝑀 → 𝐾 конструктивна, если существует алгоритм, который ∀𝑥 ∈ 𝑀
вычисляет значение 𝐹 (𝑥). Ниже под функциями мы будем понимать только

конструктивные функции.

Понятие алгоритма является начальным (неопределяемым), как, напри-

мер, понятия точки и множества в геометрии.

Рассмотрим свободный моноид 𝐺, порождённый некоторым множеством 𝑀 .

Пусть 𝑁 – некоторое множество. Функция 𝑓 : 𝐺 → 𝑁 называется индуктив-

ной, если её значение на последовательности 𝑥1 . . . 𝑥𝑛 можно вычислить по

её значению на последовательности 𝑥1 . . . 𝑥𝑛−1 и по 𝑥𝑛, то есть существует

функция 𝐹 : 𝑁 ×𝐺 → 𝑁 , для которой

𝑓(𝑥1 . . . 𝑥𝑛) = 𝐹 (𝑓(𝑥1 . . . 𝑥𝑛−1), 𝑥𝑛)

Отметим, что 1) 𝑓 есть функция от одного аргумента, который является ко-

нечной последовательностью, и 2) длина 𝑛 этой последовательности не дана,

хотя и дан её последний элемент 𝑥𝑛 (если он существует).

Таким образом, мы получаем конкретный алгоритм вычисления значений

функции 𝑓 , если известно её значение 𝑓() на пустой последовательности.

Например,

𝑛∑︁

𝑖=1

𝑥𝑖 – сумма всех членов конечной числовой последовательно-

сти является индуктивной функцией. Мы считаем, что сумма членов пустой

последовательности равна 0.

Не все функции 𝑓 : 𝐺 → 𝑁 являются индуктивными.

Индуктивным расширением функции 𝑓 : 𝐺 → 𝑁 называется такая индук-

тивная функция 𝑔 : 𝐺 → 𝐾, что существует функция 𝑄 : 𝐾 → 𝑁 такая,

что

𝑓(𝑥1 . . . 𝑥𝑛) = 𝑄(𝑔(𝑥1 . . . 𝑥𝑛))

Мы снова получаем конкретный алгоритм вычисления значений функции 𝑓 .

Задача 1. Определить, являются ли следующие функции индуктивными,

доопределив их, если необходимо, естественным образом на пустой последова-

тельности. Указать индуктивные расширения для неиндуктивных функций:

а). Функция, сопоставляющая данной последовательности 𝑥1 . . . 𝑥𝑛 перевёр-

нутую последовательность: 𝑥𝑛 . . . 𝑥1.

б). Дана последовательность 𝑦1 . . . 𝑦𝑘 десятичных цифр, где 𝑦1 ̸= 0, если
𝑘 > 1. Функция, сопоставляющая последовательности 𝑥1 . . . 𝑥𝑛 де-

сятичных цифр, где 𝑥1 ̸= 0 (если 𝑛 > 1), произведение целых чисел

𝑥1 . . . 𝑥𝑛 * 𝑦1 . . . 𝑦𝑘.
в). Среднее арифметическое последовательности целых чисел.

г). Число элементов последовательности целых чисел, равных её максималь-

ному элементу.

д). Максимальная длина монотонного (неубывающего или невозрастающего)

отрезка из идущих подряд элементов в последовательности целых чисел.

е). Максимальная длина возрастающей подпоследовательности данной после-

довательности целых чисел.

2.2. Примеры программ на языке Рефал. Языком программирования

𝐿 называется тройка {𝐷,𝑃, 𝑆}, где 𝐷 – множество данных, 𝑃 – множество про-

грамм и функция семантики 𝑆 : 𝑃 ×𝐷 → 𝐷⊥ (где ⊥ /∈ 𝐷 и 𝐷⊥ ::= 𝐷 ∪ {⊥}),
описывающая алгоритм вычисления конкретной программы 𝑝 ∈ 𝑃 на конкрет-

ном данном 𝑑 ∈ 𝐷, то есть 𝑆(𝑝, 𝑑) = 𝑝(𝑑). Причём 𝑝(𝑑) = ⊥ тогда и только

тогда, когда 𝑝 попадает в состояние аварийной остановки или в бесконечный

цикл.

Пусть фиксирован некоторый алфавит 𝒜, элементы которого мы будем на-

зывать символами. Множество данных Рефала 𝐷 может быть определено ин-

дуктивно:

∙ если d1 ∈ 𝐷, тогда (d1) ∈ 𝐷;

∙ если d1 ∈ 𝐷 и d2 ∈ 𝐷, тогда d1 d2 ∈ 𝐷;

∙ пустая последовательность является данным;

∙ ∀𝑎 ∈ 𝒜 ⇒ 𝑎 ∈ 𝐷.

Подобные определения принято кратко записывать так:

d ::= d1 d2 | (d1) | 𝑎 | empty

empty ::=

, где знак «::=» читается «по определению равно», а знак «|» есть логическая

связка «или».

Алфавит Рефала (множество символов) мы будем описывать по ходу де-

ла – постепенно расширяя его. Данные Рефала принято называть объектны-

ми выражениями. Объектным термом Рефала называется символ или объ-

ектное выражение, заключённое в скобки. Таким образом, любое объектное



126 А.П. НЕМЫТЫХ

Рефал-выражение есть конечная последовательность объектных Рефал-термов

(включая пустую последовательность).

Любой читабельный символ, который позволяет ввести клавиатура компью-

тера, заключённый в одинарные кавычки, есть символ Рефала. Например:

’;’, ’ ’, ’b’. Исключение составляет сама одинарная кавычка; её необходимо

экранировать: ’\’’.

Слово называется палиндромом, если европейское прочтение этого слова

(слева направо) совпадает с его арабским прочтением (справа налево). Сло-

во «палиндром» в переводе с греческого языка означает бегущий вспять.

Предикатом на множестве 𝑀 называется функция 𝜙 со значениями в двух-

элементном множестве 𝜙 : 𝑀 → {’T’, ’F’}. Здесь ’T’ означает истину (True),
а ’F’ – ложь (False).

Ниже дано рекурсивное определение на Рефале предиката на множестве

слов: его значение на данном конкретном слове-аргументе 𝑤0 равно ’T’ то-

гда и только тогда, когда это слово есть палиндром.

Palindrome {

= ’T’;

s.x = ’T’;

s.x e.middle s.x = <Palindrome e.middle>;

e.word = ’F’;

}

Третье предложение сравнивает на равенство первую и последнюю буквы

слова 𝑤0 и, если они совпадают, 𝑤0 есть палиндром тогда и только тогда, ко-

гда палиндромом является слово, полученное из 𝑤0 отбрасыванием первой и

последней буквы. Первые два предложения описывают базисные случаи ин-

дукции: первое – для слов чётной длины, второе – для слов нечётной длины.

Процесс проверки заканчивается, так как на каждом шаге длина слова умень-

шается и она ограничена снизу нулём.

Задача 2. Определить предикат палиндрома на языке Haskell и сравнить

это определение с данным выше.

Понятие рекурсии тесно связано с понятием математической индукции.

Рассмотрим рекурсивное определение функции, переворачивающей данную

последовательность термов. О.Д.З. переменной типа «t.» есть множество объ-

ектных Рефал-термов.

Reverse {

= ;

t.x e.rest = <Reverse e.rest> t.x;

}

Мы 1) переворачиваем последовательность, полученную из данной последо-

вательности отбрасыванием первого элемента, и 2) приписываем к результату

переворачивания сзади этот первый терм-элемент. Длина сокращённой после-

довательности на единицу меньше длины исходной, следовательно, повторяя

процесс шаг за шагом, мы придём к пустой последовательности, результат пе-

реворачивания которой есть она сама.

Задача 3. Определить аналогичную функцию на языке Haskell и сравнить

это определение с данным выше.

Задача 4. Привести пример программы, которая работает бесконечное вре-

мя.

Решим на Рефале следующую задачу. Кузнечик прыгает по прямой линии

на один метр, каждый раз случайно выбирая направление прыжка (вправо

или влево). Будем наблюдать за ним, записывая его выбор последовательно-

стью букв ’L’ (влево) и ’R’ (вправо). Определить, на каком расстоянии от

начальной позиции будет находиться кузнечик, когда он остановится.

Если в последовательности прыжков есть рядом стоящие символы ’LR’ или

’RL’, то они не влияют на ответ и их можно выкинуть из последовательности

– далее повторить сокращения, в противном случае вся последовательность

состоит из одинаковых букв; их число есть расстояние кузнечика от стартовой

позиции – вправо или влево.

Буквальное повторение этого предписания на Рефале выглядит так:

Jump {

e.1 ’LR’ e.2 = <Jump e.1 e.2>;

e.1 ’RL’ e.2 = <Jump e.1 e.2>;

e.1 = e.1;

}

Здесь в качестве имён переменных мы использовали числа. Отметим, что

второе предложение можно переставить выше первого – определение останется

правильным. Третье же предложение переставить нельзя.

Задача 5. Решить задачу про кузнечика на языке Haskell и сравнить это

решение с данным выше.

2.2.1. Как выполнить программу на языке Рефал. Пусть дана программа

𝑝. Функция семантики 𝑆(𝑝, 𝑑) = 𝑝(𝑑), по определению, может быть вычислена
только тогда, когда фиксированы оба её аргумента. Программу мы уже фик-

сировали. Данное – второй аргумент функции семантики в Рефале – можно

задать, например, посредством Рефал-определения с выделенным именем Go;

именно с этого определения алгоритм, реализующий функцию семантики, на-

чинает рассматривать Рефал-программу (принято говорить, что Go является

входной точкой программы).

Чтобы подчеркнуть, что мы будем общаться с программой именно через эту

функцию – явно укажем это в синтаксисе ключевым словом $ENTRY:

$ENTRY Go {

= <Palindrome ’долог голод’>;

}

Мы зафиксировали аргумент предиката Palindrome, и теперь его значение

может быть вычислено на вычислительной машине, но результат вычисления

для нас останется неизвестным. Чтобы компьютер напечатал результат на

экране, необходимо попросить его об этом:

$ENTRY Go {

= <Prout <Palindrome ’долог голод’>>;

}

Здесь в правой части предложения стоит композиция двух действий: вычис-

лить значение Palindrome и напечатать результат вычисления. В результате

исполнения этой программы на экране компьютера появится строка: T.

2.3. Домашнее задание.

Задача 6. Определить, являются ли следующие функции индуктивными,

доопределив их, если необходимо, естественным образом на пустой последова-

тельности. Указать индуктивные расширения для неиндуктивных функций.

Дать определение этих функций на Рефале.

а). Максимальный отрезок идущих подряд одинаковых элементов последова-

тельности.

б). Даны две последовательности 𝑥1 . . . 𝑥𝑛 и 𝑦1 . . . 𝑦𝑘. Функция для всех

𝑚 6 𝑛 сопоставляющая последовательности 𝑥1 . . . 𝑥𝑚 максимальное

число 𝑗 6 𝑘 такое, что последовательность 𝑦1 . . . 𝑦𝑗 есть подпоследова-

тельность последовательности 𝑥1 . . . 𝑥𝑚.

Задача 7. Будем все слова записывать только строчными буквами. Обоб-

щённым палиндромом назовём фразу (или последовательность фраз), кото-

рая становится палиндромом, если из неё выкинуть все знаки препинания и

пробелы. Определить на Рефале предикат на множестве текстов, множество

истинности которого совпадает с множеством обобщённых палиндромов.

Задача 8. Даны текст и строка. Определить на Рефале функцию, значение

которой есть отрезок данного текста, непосредственно следующий за первым

вхождением данной строки, если такое вхождение существует, и сам текст,

если данная строка в него не входит.

Задача 9. Даны текст и строка. Определить на Рефале функцию, значение

которой есть отрезок данного текста, непосредственно следующий за последним

вхождением данной строки, если такое вхождение существует, и саму строку,

если она в данный текст не входит.

Задача 10. В унарной системе счисления натуральные числа представля-

ются счётными палочками. Например, ’III’ = 3. Определить на Рефале

функции унарной арифметики: сложения, вычитания, умножения и деления с

остатком двух натуральных чисел.

§ 3. Унарный конструктор

построения дерева. Рефал-машина.

(Лекция №3)

Рефал – строгий язык первого порядка – имеет два неизоморфных конструк-

тора; один из которых – приписывание – ассоциативен, а другой – унарный –

используется для построения структуры произвольного дерева. Нас будет инте-

ресовать базисный Рефал (подмножество Рефала), в котором процесс вычисле-

ния программы может быть представлен в виде композиции простых действий,

называемых шагами Рефал-машины.

Язык программирования называется строгим (или аппликативным), ес-

ли значение функции 𝐹 (𝑥, . . . , 𝑦) не вычисляется, пока не вычислены все её

аргументы 𝑥, . . . , 𝑦. Примером нестрогого («ленивого») языка является язык

программирования Haskell.

Задача 1.

а). Приведите пример программы 𝑝 на языке Haskell, формальный перевод

которой на Рефал (то есть изменяется только синтаксис, но не изменя-

ется структура программы) существенно повышает её эффективность:

позволяет вычислить значение 𝑝(𝑑0) существенно быстрее и для этого

вычисления понадобится существенно меньше оперативной памяти вы-

числительной машины.

б). Приведите пример программы 𝑝 на языке Рефал, формальный перевод

которой на Haskell существенно повышает её эффективность.

3.1. Парные скобки как унарный конструктор построения дерева.

Унарный конструктор парных круглых скобок (data0) является средством

структурирования данных. Удобно думать о действии этого конструктора, как

о помещении его аргумента (внутренности скобок) внутрь матрёшки: мы мо-

жем переносить матрёшку из одного места в другое как единое целое – не

заботясь о том, что спрятано внутри неё; как и содержимое матрёшки, внут-

ренность скобки имеет конечную глубину.

Хотя формально в Рефале все функции, по определению, одного аргумента

(унарны) – конечной последовательности, конструктор скобок позволяет ими-

тировать произвольное число аргументов. Например, функцию F от двух ар-

гументов можно синтаксически представить так

<F (𝑥1 . . . 𝑥𝑛) (𝑦1 . . . 𝑦𝑚)>

или так

<F 𝑥1 . . . 𝑥𝑛 (𝑦1 . . . 𝑦𝑚)>

Функция сложения двух натуральных чисел, представленных в унарной си-

стеме счисления, может быть определена так:

Summa {

(e.n) (e.m) = e.n e.m;

}

Аналогичным образом можно имитировать отображение из 𝐷 в 𝐷 ×𝐷:

F {

. . . . . . . . . . . .

. . . = (𝑥1 . . . 𝑥𝑛) (𝑦1 . . . 𝑦𝑚);

}

Как мы знаем, каждое Рефал-данное представляет лес корневых ориенти-

рованных деревьев.

Задача 2. Определить на Рефале функцию, которая по данному лесу:

а). строит последовательность всех листьев этого леса, не изменяя порядок

их следования;

б). строит последовательность всех листьев, пропуская повторные вхождения

равных листьев, но не изменяя порядок их следования.

Расширим алфавит Рефал-символов: каждое слово, написанное в латинице,

будем считать символом, если это слово не заключено в одинарные кавычки.

Например,

∙ refal – символ языка Рефал, то есть входит в О.Д.З. s-переменной;

∙ ’refal’ – последовательность из пяти символов.

Каждое слово, состоящее из букв Кириллицы и латиницы, заключённое в

двойные кавычки, есть, по определению, Рефал-символ. Причём, ’b’ ̸= b, но

"b" = b.

После нашего расширения алфавита Рефала, следующее стихотворение Ва-

лерия Брюсова будет обобщенным палиндромом (см. предыдущую лекцию

№2).

"жестоко" "раздумье" ’.’ "ночное" "молчанье"

"качает" "виденья" "былого" ’,’

"мерцанье" "встречает" "улыбки" "сурово" ’,’

"страданье"

"глубоко" ’-’ "глубоко" ’!’

"страданье" "сурово" "улыбки" "встречает" ’...’

"мерцанье" "былого" "виденья" "качает" ’...’

"молчанье" ’.’ "ночное" "раздумье" "жестоко" ’.’

Задача 3. Дано объектное Рефал-выражение 𝑑. Написать программу, за-

меняющую в 𝑑 каждое слово Haskell на слово Refal.

Скелетом Рефал-выражения называется его скобочная структура. Напри-

мер, скелет выражения ((refal) (()) ("выражение")) есть (() (()) ()).

Задача 4. По данному объектному Рефал-выражению построить его ске-

лет.

3.2. Абстрактная Рефал-машина. В базисном подмножестве Рефала

программы представляют собой множество определений-функций. Каждое

Рефал-определение есть непустая последовательность предложений, которые

разделяются точкой с запятой. Каждое предложение имеет вид:

образец = правая_часть ;

Образец и правую часть определим индуктивно посредством формальной

грамматики:

образец ::= терм-образец образец1 | empty

терм-образец ::= (образец) | СИМВОЛ | переменная

переменная ::= e.NAME | t.NAME | s.NAME

empty ::=

правая_часть ::= терм-правая_часть правая_часть1 | empty

терм-правая_часть ::= вызов_функции | (правая_часть) | СИМВОЛ

| переменная

вызов_функции ::= <NAME аргументы>

аргументы ::= правая_часть

Здесь базисные понятия мы написали прописными буквами. Существует до-

полнительное требование: множество переменных правой части предложения

является подмножеством переменных образца этого предложения.

Все ранее рассмотренные нами программы принадлежат базисному Рефалу.

Рефал-машина реализует функцию семантики языка Рефал. Опишем аб-

страктную машину базисного Рефала. (Повсюду ниже, если не оговаривается

иное, мы будем под словом Рефал иметь в виду базисный Рефал.)

Рабочая память Рефал-машины называется полем зрения.

Перед началом работы в поле зрения помещается вызов функции Go:

<Go >

Процесс работы Рефал-машины есть исполнение конечной последовательно-

сти её шагов.

Шагом Рефал-машины называется следующая последовательность дей-

ствий:

1. Выбираем в поле зрения активный вызов функции <F d0> – алгоритм

этого выбора мы дадим ниже (стартовый активный вызов есть <Go >);

i := 1.

2. Пусть «образец𝑖 = правая_часть𝑖» есть 𝑖-тое предложение функции F,

тогда решается уравнение «образец𝑖 = d0», если это уравнение несов-

местно, то переходим к решению уравнения «образец(𝑖+1) = d0». Если

же 𝑖-ое предложение последнее, то значение функции F на аргументе d0
не определено и Рефал-машина переходит в состояние аварийной оста-

новки.

3. Пусть уравнение «образец𝑖 = d0» имеет решение. Если решение един-

ственно, тогда переходим к пункту 4, иначе существует e-переменная, ко-

торая принимает более одного значения. Такие e-переменные называют-

ся открытыми в данном образце. Рассмотрим самую левую открытую

переменную e.y; среди всех возможных значений e.y – конечных после-

довательностей 𝑥0
1 . . . 𝑥0

𝑛 выбираем последовательность с наименьшей

длиной 𝑛 и считаем только её значением переменной e.y, отбрасывая все

решения уравнения «образец𝑖 = d0», которые не удовлетворяют данно-

му условию. Таким образом, мы сузили множество решений 𝑀 . Если в

образце остались другие открытые переменные, то выбираем среди них

самую левую и выбираем из множества её значений самое короткое и

т. д., пока в образце не останется открытых переменных.

4. Имеем выбранное предложение «образец𝑖 = правая_часть𝑖» и выбран-

ное нами его единственное решение. Делаем подстановку значений пе-

ременных в правую часть 𝑖-ого предложения и заменяем в поле зрения

активный вызов <F d0> результатом этой подстановки.

5. Если в изменённом поле зрения нет вызовов функций, то машина оста-

навливается, и поле зрения считается результатом её работы.

Заметим, что реальный (не абстрактный) алгоритм шага Рефал-машины

может быть иным. Например, в Рефале-5 пункты 2 и 3 шага делаются одно-

временно и ищется сразу необходимое решение уравнения, а не всё множество

решений.

Поле зрения, по определению, есть

поле_зрения ::= терм-поле_зрения поле_зрения1 | empty

терм-поле_зрения ::= <NAME поле_зрения> | (поле_зрения) | СИМВОЛ

Активным вызовом функции в поле зрения называется вызов, правая за-

крывающая угловая скобка которого является самой левой из всех правых уг-

ловых скобок.

Задача 5. Дана последовательность, элементами которой являются скоб-

ки ’(’, ’)’, ’<’, ’>’. Написать программу, выдающую сообщение об ошибке,

если существует скобочный уровень одного типа скобок, на котором нарушена

правильность скобочной структуры любого типа скобок. Если же на вход про-

граммы подана правильная скобочная структура, то преобразовать каждую

пару ’(’ . . . ’)’ в структурные скобки ( . . . ), а каждую пару ’<’ . . . ’>’ в

(Call . . . ). Например,

’(<()>)’ ⇒ ( (Call ()) ).

3.3. Домашнее задание.

Задача 6. Скобочной унарной системой счисления назовём систему счис-

ления, в которой натуральные числа представляются парными вложенными

скобками. Например, ((())) = 3. Определить на Рефале функции скобочной

унарной арифметики: сложения, вычитания, умножения и деления с остатком

двух натуральных чисел.

Задача 7. Дана правильная скобочная структура 𝑑, содержащая только

скобки. Требуется написать программу, результатом которой является после-

довательность всех различных объектных Рефал-выражений, скелеты которых

совпадают с 𝑑 и которые содержат ровно один символ b. Например,

(()) () ⇒ ’[’ b (()) () ’]’ ’[’ (b ()) () ’]’ ’[’ ((b)) () ’]’

’[’ (() b) () ’]’ ’[’ (()) b () ’]’ ’[’ (()) (b) ’]’

’[’ (()) () b ’]’

Задача 8. Определить, может ли быть получен палиндром из данного сло-

ва путём перестановки букв.

Задача 9. (Примитивный корень последовательности). На входе конеч-

ная последовательность объектных Рефал-термов. Найти такую последова-

тельность минимальной длины, повторением которой несколько раз можно по-

лучить входную последовательность.

Задача 10. Анаграмма – литературный прием, состоящий в перестановке

букв определенного слова (или словосочетания), что в результате дает другое

слово или словосочетание. Преобразовать текст в анаграмму, где одинаковые

символы стоят подряд (так шифровали открытия средневековые ученые).

§ 4. Арифметика

(Лекция №4)

Целочисленная арифметика в Рефале является точной. Другими словами:

ограничение на разрядность (длину) целых чисел отсутствует. Представление

целых чисел зависит от конкретного диалекта Рефала. Мы рассмотрим, каким

образом целые числа представлены в Рефале-5.

4.1. Представление целых чисел. Прежде всего, снова расширим ал-

фавит Рефал-символов: натуральное число 𝑛 принадлежит алфавиту тогда и

только тогда, когда

0 6 𝑛 < 232

Целые числа в Рефале представляются в системе счисления по основанию 232.
Таким образом, по определению, целое число принадлежит О.Д.З. s-пере-

менной тогда и только тогда, когда оно является цифрой в этой системе счисле-

ния. Чтобы избежать путаницы с десятичными цифрами, мы будем называть

цифры в системе счисления по основанию 232 макро-цифрами.

Задача 1. Используя школьные алгоритмы, определить на Рефале функ-

ции арифметики в двоичной системе счисления: сложения, вычитания, умно-

жения и деления с остатком двух натуральных чисел.

Как и в десятичной системе счисления, произвольное натуральное число есть

последовательность макро-цифр. Например,

232 = 1 0
232 − 1 = 4294967295

Заметим, что ’1’ ̸= 1 и ’4294967295’ ̸= 4294967295 .

Отрицательные числа начинаются с Рефал-символа ’-’. Например,

−(232 + 1) = ’-’ 1 1 .

Аналогично разрешается приписывать символ ’+’ перед неотрицательным

числом.

4.2. Встроенные функции арифметики. Некоторые функции суще-

ствуют в Рефал-машине изначально – они называются встроенными.

Все встроенные функции арифметики целых чисел (сложение – Add (или

+), вычитание – Sub (или -), умножение – Mul (или *), деление – Div (или /),

деление с остатком – Divmod, остаток от деления – Mod, сравнение – Compare)

логически имеют два аргумента, первый из которых заключён в структурные

скобки. Например,

<Add (1 0) 1> = 1 1

Результат деления с остатком есть пара:

<Divmod (e.n) e.m> = (e.q) e.r

где e.n = e.q*e.m + e.r, 0 6 |e.r| < |e.m| и знак остатка e.r совпадает со

знаком делимого e.n.

Задача 2. Дана последовательность цифр натурального числа 𝑛, данного
в системе счисления по основанию 232. Построить последовательность цифр

числа 𝑛 в десятичной системе счисления.

Функция сравнения двух целых чисел Compare возвращает: ’+’, если первое

больше второго; ’-’, если первое меньше второго, и ’0’, если числа равны.

В том случае, когда не возникает неоднозначности, структурные скобки, за-

ключающие первый аргумент встроенных функций арифметики, можно опус-

кать.

Задача 3. Реализовать на Рефале алгоритм умножения столбиком двух на-

туральных чисел, данных в в системе счисления по основанию 232. Встроенные
функции разрешается использовать только тогда, когда их аргументы являют-

ся макро-цифрами.

4.3. Два типа рекурсии. На примере функции 𝑛! покажем два вида ре-

курсии.

Рекурсия первого вида соответствует умножению чисел от 1 до 𝑛:

𝑛! = 1 * . . . * 𝑛.
Fact {

1 = 1;

s.n = <Mul (s.n) <Fact <Sub s.n 1>>>;

}

Задача 4. Объясните, почему аргументом функции Fact выбрана макро-

цифра, а не произвольное натуральное число. Как будет работать Рефал-

машина при попытке вычислить вызов <Fact 0>?

Рекурсия второго вида соответствует умножению чисел в обратном порядке:

𝑛! = 𝑛 * . . . * 1.

Fact1 {

s.n = <Fact2 s.n 1>;

}

Fact2 {

s.n s.n = s.n;

s.n s.m = <Mul (s.m) <Fact2 s.n <Add s.m 1>>>;

}

По способу завершения рекурсии, рекурсию первого вида назовём рекурсией

с выходом по исчерпанию; рекурсию второго вида – рекурсией с выходом по

наполнению.

Задача 5. Дано натуральное число 𝑛. Вычислить на Рефале 𝑛-ое число

Каталана 𝑐𝑛 (см. лекцию 1, задачу 7).

4.4. Фибоначчиева система счисления. Числа Фибоначчи определяют-

ся индуктивно: 𝑓1 = 𝑓2 = 1, 𝑓𝑛+2 = 𝑓𝑛 + 𝑓𝑛+1.

Задача 6. Дано натуральное число 𝑛. Вычислить на Рефале 𝑛-ое число

Фибоначчи 𝑓𝑛.

Докажем, что любое натуральное число 𝑚 > 0 можно однозначным образом

представить в виде: 𝑚 = 𝑎𝑛 * 𝑓𝑛 + . . . + 𝑎2 * 𝑓2, где для всех 𝑖 < 𝑛 верно, что

𝑎𝑖 равно нулю или единице и 𝑎𝑛 = 1, для всех 𝑖 верно 𝑎𝑖 * 𝑎𝑖+1 = 0. Таким

образом, получаем фибоначчиеву систему счисления: 𝑎𝑖 – цифры числа 𝑚 в

этой системе счисления.

Доказательство будет конструктивно и индуктивно.

Вычтем из 𝑚0 = 𝑚 наибольшее из не превосходящих его чисел Фиббоначи

𝑓𝑛 (с наибольшим номером) и рассмотрим 𝑚1 = 𝑚0 − 𝑓𝑛, 𝑎𝑛 = 1.

Предположим, что 𝑘 шагов уже выполнены, и мы имеем некоторое число

𝑚𝑘 и последовательность фибоначчиевых цифр 𝑎𝑛, . . . , 𝑎𝑛−𝑘, состоящую из

нулей и единиц. Тогда на 𝑘 + 1 шаге построим фибоначчиеву цифру 𝑎𝑛−(𝑘+1)

и число 𝑚𝑘+1.

Если 𝑚𝑘 < 𝑓𝑛−𝑘, то 𝑎𝑛−(𝑘+1) := 0 и 𝑚𝑘+1 := 𝑚𝑘,

иначе 𝑎𝑘+1 := 1 и 𝑚𝑘+1 := 𝑚𝑘 − 𝑓𝑛−𝑘.

Покажем корректность приведённого алгоритма.

На каждом шаге 𝑘 уменьшается номер 𝑛−𝑘 числа Фибоначчи, сравниваемого

с 𝑚𝑘 > 0 и, следовательно, уменьшается 𝑓𝑛−𝑘.

Индукцией по 𝑘 покажем, что ∀𝑘.(𝑚𝑘 > 𝑓𝑛−𝑘) ⇒ (𝑚𝑘+1 < 𝑓𝑛−(𝑘+1)).

1. Базис индукции: 𝑓𝑛+1 > 𝑚0 > 𝑓𝑛 ⇒ 𝑚1 = (𝑚0−𝑓𝑛) < (𝑓𝑛+1−𝑓𝑛) = 𝑓𝑛−1.

Базис индукции доказан.

2. Шаг индукции: пусть утверждение верно ∀𝑖 6 𝑘, тогда либо

𝑚𝑘+1 < 𝑓𝑛−(𝑘+1) и посылка нашего утверждения для 𝑘 + 1 ложна; либо

𝑚𝑘+1 > 𝑓𝑛−(𝑘+1) и тогда, по определению алгоритма, 𝑚𝑘+2 = 𝑚𝑘+1 −
𝑓𝑛−(𝑘+1).

Кроме того, по предположению индукции, неравенство 𝑚𝑘+1 > 𝑓𝑛−(𝑘+1)

влечёт 𝑚𝑘 < 𝑓𝑛−𝑘 и, следовательно, 𝑚𝑘+1 = 𝑚𝑘. Таким образом,

𝑚𝑘+2 = 𝑚𝑘+1−𝑓𝑛−(𝑘+1) = (𝑚𝑘−𝑓𝑛−(𝑘+1)) < (𝑓𝑛−𝑘−𝑓𝑛−(𝑘+1)) = 𝑓𝑛−(𝑘+2).

Шаг индукции доказан.

Следствие 1. Алгоритм разложения натурального числа к виду

𝑎𝑛*𝑓𝑛+. . .+𝑎2*𝑓2, где ∀𝑖 < 𝑛. 𝑎𝑖 равно нулю или единице и 𝑎𝑛 = 1, заканчивает

свою работу за конечное число шагов.

Действительно, так как 𝑓𝑛−(𝑘+1) 6 𝑓𝑛−𝑘 и min(𝑚𝑘,𝑚𝑘+1) = 𝑚𝑘+1, то дока-

занное утверждение даёт 0 6 𝑚𝑘+1 < 𝑓𝑛−𝑘, а последовательность 𝑓𝑛−𝑘 при

𝑘 < 𝑛− 1 строго убывает и 𝑓𝑛−(𝑛−2) = 𝑓2 = 1.

Пример:

𝑚 = 𝑚0 = 19,𝑚1 = 19 − 13 = 19 − 𝑓7 = 6; 𝑎7 = 1;
𝑚1 < 𝑓6 = 8 ⇒ 𝑎6 = 0;𝑚2 = 𝑚1;

𝑚2 > 𝑓5 = 5 ⇒ 𝑎5 = 1;𝑚3 = 𝑚2 − 𝑓5 = 6 − 5 = 1;
𝑚3 < 𝑓4 = 3 ⇒ 𝑎4 = 0;𝑚4 = 𝑚3;
𝑚4 < 𝑓3 = 2 ⇒ 𝑎3 = 0;𝑚5 = 𝑚4;

𝑚5 > 𝑓2 = 1 ⇒ 𝑎2 = 1;𝑚6 = 𝑚5 − 𝑓2 = 1 − 1 = 0;

и имеем 1910 = 101001𝑓 = 𝑓7 + 𝑓5 + 𝑓2 = 1310 + 510 + 110.

Задача 7. Напишите программу:

а). переводящую числа из фибоначчиевой системы счисления в десятичную;

б). переводящую числа из десятичной системы счисления в фибоначчиеву;

в). сложения двух натуральных чисел в фибоначчиевой системе счисления;

г). умножения двух натуральных чисел в фибоначчиевой системе счисления.

4.5. Домашнее задание.

Задача 8. В шестнадцатеричной системе счисления цифры представлены

символами: ’0’, ’1’, ’2’, ’3’, ’4’, ’5’, ’6’, ’7’, ’8’, ’9’, ’A’, ’B’, ’C’,

’D’, ’E’, ’F’. Используя школьные алгоритмы, определить на Рефале функции

арифметики в шестнадцатеричной системе счисления: сложения, вычитания,

умножения и деления с остатком двух натуральных чисел.

Задача 9. Написать программу, вычисляющую наибольший общий дели-

тель двух натуральных чисел методом Евклида.

Задача 10. Написать программу, вычисляющую наименьшее общее крат-

ное двух натуральных чисел.

Задача 11. Придумать представление рациональных чисел посредством

данных Рефала. Реализовать рациональную арифметику (функции сложения,

вычитания, умножения, деления двух рациональных чисел).

Задача 12. Дано натуральное число 𝑛. Вычислить на Рефале 𝑛-ое чис-

ло Фибоначчи 𝑓𝑛, если требуется, чтобы число арифметических операций при

этом вычислении было пропорционально log 𝑛.

§ 5. Функциональность. Ввод-вывод.

(Лекция №5)

Второе определяющее слово реФала есть функциональный. Язык програм-

мирования {𝐷,𝑃, 𝑆} называется функциональным, если для всех 𝑝0 ∈ 𝑃
𝑆(𝑝0, 𝑑) есть функция из 𝐷 в 𝐷⊥ (см. лекцию №2). Другими словами, про-

грамма 𝑝0 представляет собой определение некоторого отображения из 𝐷 в

𝐷⊥.
Функциональным языком, в строгом смысле приведённого выше определе-

ния, может быть только чисто теоретический язык программирования.

Задача 1. Какие из рассмотренных в предыдущих семинарах программы

не являются определениями функций из 𝐷 в 𝐷⊥, где 𝐷 – множество данных

Рефала?

Причина состоит в том, что в реальных языках программирования всегда

существуют программы с побочными действиями. И без таких программ

мы не сможем узнать, что же вычислил компьютер, и вычислил ли вообще

что-нибудь. Мы рассмотрим некоторые из подобных программ ниже.

Среди реальных (не чисто теоретических) языков программирования функ-

циональными принято называть те языки, в которых число выразительных

синтаксических средств, позволяющих конструировать программы с побочны-

ми действиями, сведено до минимума. Типичным свойством функционального

языка программирования является отсутствие в нём глобальных переменных.

Задача 2. Приведите пример языка программирования, не являющегося

функциональным.

5.1. Встроенный ввод-вывод. Мы уже использовали встроенную «функ-

цию» печати Prout: её значением всегда является пустое выражение; а побоч-

ным действием – вывод на терминал своего аргумента.

Часто бывает необходимо вывести результат вычислений Рефал-машины в

файл: с целью длительного хранения этого результата, другой причиной может

быть большой объём выводимых данных.

Организация работы Рефал-машины с файлами зависит от конкретного диа-

лекта Рефала. Мы будем интересоваться Рефалом-5.

File, в переводе с английского языка, – папка бумаг; она может быть пустая

или содержать некоторую текстовую информацию. На обложке папки написа-

но её имя.

Чтобы работать с содержимым файла, его, как и папку, нужно сначала

открыть. При завершении работы с содержимым файла, файл необходимо

закрыть. Цель работы может быть разная: чтение содержимого файла; до-

бавление текста; уничтожение всей информации как устаревшей и запись в

файл новой информации. У папок, взятых с разных книжных полок, могут

быть совпадающие имена; чтобы не возникала путаница при работе с ними,

принято прикреплять к ним (в момент открытия) уникальные этикетки, ко-

торые являются макро-цифрами – описателями (дескрипторами – descriptors).

Посредством этих описателей и происходит реальная работа с файлами.

Файл с именем ’name’ открывает вызов

<Open s.Mode s.D ’name’>

где s.D – дескриптор файла, который задаёт программист, а s.Mode указывает

цель работы с содержимым файла: ’r’ – открыть для чтения, ’a’ – открыть

для дополнения, ’w’ – открыть для записи, предварительно уничтожив его

содержимое. Если происходит попытка открыть для записи несуществующий

файл, то он автоматически создаётся.

Закрываем файл посредством вызова

<Close s.D>

Значения «функций» Open и Close суть пустые выражения.

«Функция» Putout записывает (в виде одной строки) значение e.Expr в

файл, связанный с дескриптором s.D:

<Putout s.D e.Expr>

Следующий вызов этой «функции» будет производить запись с начала следу-

ющей строки.

Задача 3. (Cоревнование) Написать как можно более короткую программу,

записывающую в файл turnip.txt русскую народную сказку «Репка».

Аналогично выводу, чтение (ввод) в Рефале-5 также построчный: 𝑛-ый вы-

зов

<Get s.D>

выдаёт 𝑛-ую строку файла, связанного с s.D.

Чтение с клавиатуры производится вызовом

<Card >

который завершает чтение по нажатию клавиши Enter.

Задача 4. Написать программу, переставляющую строки в данном файле

в обратном порядке.

Значение вызова <Arg s.n> есть s.n-ый аргумент командной строки опера-

ционной системы при запуске Рефал-машины.

Задача 5. Написать программу, значением которой является последова-

тельность всех аргументов Рефал-машины, передаваемых командной строкой

операционной системы.

5.2. Пример А.В. Корлюкова. Рассмотрим пример Александра Кор-

люкова, показывающий выразительные возможности ассоциативной конкате-

нации.

Пусть 𝐵 = {’0’, ’1’}, 𝑀 ⊃ 𝐵3. Требуется определить на Рефале функцию

𝐹 : 𝑀 → 𝐵2 такую, что ∀𝑥 ∈ 𝐵3. 𝐹 (𝑥) есть число 𝑛 единиц в 𝑥, где 𝑛 представ-

лено в двоичной системе счисления и, возможно, содержит в своём представ-

лении лидирующие нули.

Задачу начнём решать с полного перебора:

F {

’111’ = ’11’;

’110’ = ’10’;

’101’ = ’10’;

’100’ = ’01’;

’011’ = ’10’;

’010’ = ’01’;

’001’ = ’01’;

’000’ = ’00’;

}

В этом определении предложения перестановочны. Объединяя первое и по-

следнее предложение, получаем

s.d s.d s.d = s.d s.d;

Аналогично объединяем все предложения с совпадающими правыми частя-

ми. Наше определение теперь выглядит нижеследующим образом:

F {

s.d s.d s.d = s.d s.d;

e.x ’1’ e.y ’1’ e.z = ’10’;

e.x ’0’ e.y ’0’ e.z = ’01’;

}

Заметим, что во втором и в третьем предложении значения двух переменных

равны пустому выражению; значение третьей переменной во втором предложе-

нии равно ’0’, а в третьем – ’1’. Таким образом, второе и третье предложение

также можно объединить:

F {

s.d s.d s.d = s.d s.d;

e.x s.d e.y s.d e.z = s.d e.x e.y e.z;

}

Теперь нетрудно видеть, что первое предложение лишнее. Действительно, в

результате отождествления трёх равных цифр с образцом e.x s.d e.y s.d e.z:

e.x s.d e.y s.d e.z = d0 d0 d0
имеем

s.d = d0
e.x =

e.y =

e.z = d0
и правая часть второго предложения становится равной d0 d0, то есть совпа-

дает с правой частью первого предложения.

Окончательно имеем определение:

F {

e.x s.d e.y s.d e.z = s.d e.x e.y e.z;

}

В результате наших преобразований область определения функции, данной

стартовым определением, была расширена.

Задача 6. Написать программу, стирающую двойные (лишние) пробелы в

данном тексте, то есть заменяющую каждую группу стоящих подряд пробелов

единственным пробелом.

5.3. Домашнее задание.

Задача 7. Придумать представление понятия множества посредством дан-

ных Рефала. Определить функции объединения, пересечения и разности двух

множеств.

Задача 8. (Cоревнование) Написать как можно более короткую программу,

записывающую в файл Jack.txt нижеследующее стихотворение Джонатана

Свифта:

The House That Jack Built

Jonathan Swift

This is the house that Jack built.

This is the malt
That lay in the house that Jack built.

This is the rat,
That ate the malt
That lay in the house that Jack built.

This is the cat,
That killed the rat,
That ate the malt
That lay in the house that Jack built.

This is the dog,
That worried the cat,
That killed the rat,
That ate the malt

That lay in the house that Jack built.

This is the cow with the crumpled horn,
That tossed the dog,
That worried the cat,
That killed the rat,
That ate the malt
That lay in the house that Jack built.

This is the maiden all forlorn,
That milked the cow with the crumpled horn,
That tossed the dog,
That worried the cat,
That killed the rat,
That ate the malt
That lay in the house that Jack built.

This is the man all tattered and torn,
That kissed the maiden all forlorn,
That milked the cow with the crumpled horn,
That tossed the dog,
That worried the cat,
That killed the rat,
That ate the malt
That lay in the house that Jack built.

This is the priest all shaven and shorn,
That married the man all tattered and torn,
That kissed the maiden all forlorn,
That milked the cow with the crumpled horn,
That tossed the dog,
That worried the cat,
That killed the rat,
That ate the malt
That lay in the house that Jack built.

This is the cock that crowed in the morn,
That waked the priest all shaven and shorn,
That married the man all tattered and torn,
That kissed the maiden all forlorn,
That milked the cow with the crumpled horn,
That tossed the dog,
That worried the cat,
That killed the rat,
That ate the malt
That lay in the house that Jack built.

This is the farmer sowing the corn,
That kept the cock that crowed in the morn,
That waked the priest all shaven and shorn,
That married the man all tattered and torn,
That kissed the maiden all forlorn,
That milked the cow with the crumpled horn,

That tossed the dog,
That worried the cat,
That killed the rat,
That ate the malt
That lay in the house that Jack built.

Задача 9. Сонет -– стихотворение из 14 строк, которые образуют особым

образом построенную строфу. Венок сонетов состоит из 15 стихотворений, где

последняя строка каждого из 14 сонетов является началом следующего. За-

ключительный сонет воспроизводит первые строки всех 14 предыдущих соне-

тов (см. [9]).

Написать программу, проверяющую, является ли данное в файле стихотво-
рение венком сонетов (см. определение венка сонетов на странице [9]). Про-
грамму протестировать на следующем стихотворении Владимира Солоухина
(см. также файл [10]).

1

Венок сонетов – давняя мечта.
Изведать власть железного канона!
Теряя форму, гибнет красота,
А форма чётко требует закона.

Невыносима больше маета
Аморфности, неряшливости тона,
До скрежета зубовного, до стона,
Уж если так, пусть лучше немота.

Прошли, прошли Петрарки времена.
Но в прежнем ритме синяя волна
Бежит к земле из дали ураганной.

И если ты всё – мастер и поэт,
К тебе придёт классический сонет –
Вершина формы строгой и чеканной.

2

Вершина формы строгой и чеканной –
Земной цветок: жасмин, тюльпан, горлец,
Кипрей и клевер, лилии и канны,
Сирень и роза, ландыш, наконец.

Любой цветок сорви среди поляны –
Тончайшего искусства образец,
Не допустил Ваятеля резец
Ни одного малейшего изъяна.

Как скудно мы общаемся с цветами.
Меж красотой и суетными нами
Лежит тупая жирная черта.

Но не считай цветенье их напрасным,
Мы к ним идём, пречистым и прекрасным,
Когда невыносима суета.

3

Когда невыносима суета,
И возникает боль в душе глубоко,
И складка горькая ложится возле рта,
Я открываю том заветный Блока.

Звенит строка, из бронзы отлита,
Печального и гордого пророка.
Душа вольна, как дальняя дорога,
И до звезды бездонна высота.

О Блок! О Блок! Мертвею, воскреси!
Кидай на землю, мучай, возноси
Скрипичной болью, музыкой органной!

Чисты твоей поэзии ключи.
Кричать могу, молчанью научи,
К тебе я возвращаюсь в день туманный.

4

К тебе я возвращаюсь в день туманный,
О Родина, ужели это сны?
Кладу букет цветов благоуханный
На холмик глины около сосны.

И около берёзы. И в Тарханах.
И у церковной каменной стены.
Поэты спят; те стойкой ресторанной,
Те пошлостью, те пулей сражены.

А нас толпа. Мы мечемся. Мы живы.
Слова у нас то искренни, то лживы.
И без звезды живём и без креста.

Но есть дела. Они первостепенны.
Да ты ещё маячишь неизменно,
О белизна бумажного листа!

5

О белизна бумажного листа!
Ни завитка, ни чёрточки, ни знака.
Ни мысли и ни кляксы. Немота.
И слепота. Нейтральная бумага.

Пока она безбрежна и чиста,
Нужны или наивность, иль отвага
Для первого пятнающего шага –
Оставишь след и не сотрёшь следа.

Поддавшись страшной власти новизны,
Не оскверняй великой белизны
Поспешным жестом, пошлостью пространной.

Та белизна – дорога и судьба,

Та белизна – царица и раба,
Она источник жажды окаянной.

6

Она источник жажды окаянной.
Вся жизнь, что нам назначено прожить.
И соль и мёд, и горечь браги пьяной
Чем больше пьёшь, тем больше хочешь пить.

Сладко́ вино за стенкою стаканной,
Мы пьём и льём, беспечна наша прыть,
До той поры, когда уж нечем крыть
И жалок мусор мелочи карманной.

За ледоход! За дождь! За листопад!
За синий свод – награду из наград,
За жаворонка в полдень осиянный,

За все цветы, за все шипы земли,
За постоянно брезжущий вдали
Манящий образ женщины желанной!

7

Манящий образ женщины желанной . . .
Да – помыслы, да – книги, да – борьба.
Но всё равно одной улыбкой странной
Она творит героя и раба.

Ты важный, нужный, яркий, многогранный,
Поэт, главарь, – завидная судьба!
Уйдёт с другим, и ты сойдёшь с ума,
И будешь бредить пулею наганной.

Немного надо – встретиться любя.
Но если нет, то всюду ждут тебя
В пустых ночах пустые города,

Да всё-таки – надежды слабый луч,
Да всё-таки – сверкнувшая из туч
В ночи осенней яркая звезда.

8

В ночи осенней яркая звезда,
Перед тобой стою среди дороги.
О чём горишь, зовёшь меня куда,
Какие ждут невзгоды и тревоги?

Проходит лет, событий череда,
То свет в окне, то слёзы на пороге,
Глаза людей то ласковы, то строги,
Всё копится для страшного суда.

Для каждого наступит судный день:
Кем был, кем стал, где умысел, где лень?

Ты сам себе и жертва и палач.

Ну что ж, ложись на плаху головою,
Но оставайся всё-таки собою,
Себя другим в угоду не иначь.

9

Себя другим в угоду не иначь.
Они умней тебя и совершенней,
Но для твоих вопросов и задач
Им не найти ответов и решений.

Ты никуда не денешься, хоть плачь,
От прямиков, окольностей, кружений,
От дерзновенных взлётов и крушений,
От всех своих побед и неудач.

Привалов нет, каникул не бывает.
В пути не каждый сразу понимает,
Что жизнь не тульский пряник, не калач.

Рюкзак годов всё крепче режет плечи,
Но если вышел времени навстречу,
Души от ветра времени не прячь!

10

Души от ветра времени не прячь . . .
Стоять среди железного мороза
Умеет наша светлая берёза,
В огне пустынь не гибнет карагач.

Но точит волю вечная угроза.
Но подлецом не должен быть скрипач.
Но губят песню сытость, ложь и проза,
Спасти её – задача из задач.

Берёшь, глядишь: такие же слова,
Похожа на живую, а мертва.
Но если в ней сознанье угадало

Хоть уголёк горячий и живой,
Ты подними её над головой,
Чтобы её как факел раздувало.

11

Чтобы её как факел раздувало,
Ту истину, которая в тебе,
Не опускай тяжёлого забрала,
Летя навстречу буре и борьбе.

Тлен не растлил, и сила не сломала.
И медлит та, с косою на горбе.
Хвала, осанна, ода, гимн судьбе –
Ты жив и зряч, не много и не мало!

С тобой деревья, небо над тобой.
Когда же сердце переполнит боль,
Оно взорвётся ярко, как фугас.

Возможность эту помни и держи,
Для этого от сытости и лжи
Хранится в сердце мужества запас.

12

Хранится в сердце мужества запас,
Как раньше порох в крепости хранили,
Как провиант от сырости и гнили,
Как на морском судёнышке компас.

Пускай в деревьях соки отбродили,
Пусть летний полдень засуху припас,
Пусть осень дышит холодом на нас
И журавли над нами оттрубили,

Пусть на дворе по-зимнему темно,
Согреет кровь старинное вино,
Уздечкой звякнет старенький пегас.

Придут друзья – обрадуемся встрече,
На стол поставим пушкинские свечи,
Чтоб свет во тьме, как прежде, не погас!

13

И свет во тьме, как прежде, не погас.
Да разве свет когда-нибудь погаснет?!
Костром горит, окном манит в ненастье,
В словах сквозит и светится из глаз.

Пустые толки; домыслы и басни,
Что можно, глыбой мрака навалясь,
Идущий день отсрочить хоть на час,
Нет ничего смешнее и напрасней!

И мрак ползёт. То атомный распад.
То душ распад. То свист, а то поп-арт.
Приоритет не духа, а металла.

Но под пустой и жалкой суетой
Он жив, огонь поэзии святой,
И тьма его, как прежде, не объяла.

14

И тьма его, как прежде, не объяла,
Мой незаметный, робкий огонёк.
Несу его то бодро, то устало,
То обогрет людьми, то одинок.

Уже не мало сердце отстучало,
Исписан и исчеркан весь листок,

Ошибок – воз, но этот путь жесток,
И ничего нельзя начать сначала.

Не изорвать в сердцах черновика,
Не исправима каждая строка,
Не истребима каждая черта.

С рассветом в путь, в привычную дорогу.
Ну, а пока дописан, слава Богу,
Венок сонетов – давняя мечта.

15

Венок сонетов – давняя мечта,
Вершина формы строгой и чеканной,
Когда невыносима суета,
К тебе я обращаюсь в день туманный.

О белизна бумажного листа!
Она источник жажды окаянной,
Манящий образ женщины желанной,
В ночи осенней яркая звезда!

Себя другим в угоду не иначь.
Души от ветра времени не прячь,
Чтобы её, как факел, раздувало.

Хранится в сердце мужества запас.
И свет во тьме, как прежде, не погас,
И тьма его, как прежде, не объяла!

§ 6. Алгоритмическая полнота.

Алгорифмы Маркова.

(Лекция №6)

Язык рефАЛ алгоритмический. Естественно встаёт вопрос об алгоритми-

ческой полноте Рефала. Оказывается, даже базисное подмножество Рефала,

которое мы до сих пор только и рассматривали, является алгоритмически пол-

ным языком. То есть любой алгоритм может быть запрограммирован на ба-

зисном Рефале.

6.1. Алгоритмическая полнота.

Задача 1. Воспользовавшись тезисом Чёрча, докажите что базисный Ре-

фал является алгоритмически полным языком.

Бо́льшая часть языков программирования обладает свойством алгоритми-

ческой полноты. Следовательно, любую программу, написанную на одном из

таких языков программирования, можно перевести на другой такой язык.

Разнообразие языков программирования объясняется их ориентацией на

конкретные классы задач. Вопрос не в том можно или нельзя решить какую-то

задачу на данном языке программировании 𝐿, а в том удобно или нет решать

эту задачу на языке 𝐿. Язык есть инструмент для решения задачи: хотя и

можно гвозди забивать сковородкой (и даже лбом), но очень неудобно, – пред-

почитают использовать молоток.

Рефал ориентирован на преобразование текстов (символьной информации).

6.2. Алгорифмы Маркова. Операционная семантика языка Рефал ос-

нована на теоретической модели вычислений, называемой нормальными алго-

рифмами А.А. Маркова.

Пусть дан некоторый алфавит 𝒜. Рассмотрим свободную алфавитную по-

лугруппу 𝐺 над 𝒜. В качестве элементарной операции, на базе которой строят-

ся алгорифмы Маркова, используется подстановка одного элемента 𝐺 вместо

другого. Если 𝑥, 𝑦 ∈ 𝐺, то выражения 𝑥 → 𝑦 и 𝑥 → ∙ 𝑦 будем называть

формулами подстановки. При этом предполагается, что стрелка → и точка ∙
не являются буквами алфавита 𝒜.

Формула подстановки 𝑥 → 𝑦 называется простой. Формула подстановки

𝑥 → ∙ 𝑦 называется заключительной. Пусть 𝑃 → 𝑄 обозначает любую

из формул подстановки (простую или заключительную). Конечная последова-

тельность формул подстановки

⎧
⎪⎪⎪⎨
⎪⎪⎪⎩

𝑃1 → 𝑄1

𝑃2 → 𝑄2

. . . . . . . . .

𝑃𝑟 → 𝑄𝑟

называется схемой алгорифма.

Скажем, что последовательность (как элемент) 𝑥 ∈ 𝐺 входит в последова-

тельность 𝑦 ∈ 𝐺, если существуют такие (возможно, пустые) 𝑢, 𝑣 ∈ 𝐺, что
𝑦 = 𝑢 𝑥 𝑣. Длину последовательности 𝑥 ∈ 𝐺 обозначим |𝑥|.

Задача 2. Пусть дано 𝑦 ∈ 𝐺. Скажем, что существует два перекрываю-

щихся вхождения 𝑥 ∈ 𝐺 в 𝑦, если ∃𝑣 ∈ 𝐺 такое, что 𝑣 входит в 𝑦, |𝑥| < |𝑣| < 2|𝑥|
и ∃𝑝 ∈ 𝐺 ∃𝑞 ∈ 𝐺. 𝑣 = 𝑥 𝑝 = 𝑞 𝑥. Написать на Рефале программу, проверяю-

щую, что для данной последовательности 𝑦 не существует последовательности
𝑥, которая имеет перекрывающееся вхождение в 𝑦.

Работа алгоритма, порождённого схемой алгорифма, может быть описана

следующим образом. Пусть дано 𝑝 ∈ 𝐺.

∙ Находим первую в схеме алгорифма формулу подстановки 𝑃𝑚 → 𝑄𝑚

такую, что 𝑃𝑚 входит в 𝑝.

∙ Подставляем 𝑄𝑚 вместо самого левого вхождения 𝑃𝑚 в 𝑝. Пусть 𝑟1 –

результат такой подстановки.

∙ Если 𝑃𝑚 → 𝑄𝑚 – заключительная формула подстановки, то работа

алгоритма заканчивается и его значением на 𝑝 является 𝑟1.

∙ Если 𝑃𝑚 → 𝑄𝑚 – простая, то применим к 𝑟1 тот же поиск, который

только что применяли к 𝑝, и так далее.

∙ Если мы на 𝑖-ом шаге получим такое 𝑟𝑖, что ни одна из 𝑃1, . . . , 𝑃𝑟 не

входит в 𝑟𝑖, то работа алгоритма заканчивается и 𝑟𝑖 будет его значением
на 𝑝.

∙ При этом возможно, что описанный процесс никогда не закончится. В

таком случае мы скажем, что алгоритм не применим к 𝑝.

Алгоритм, определённый таким образом, называется нормальным алгориф-

мом Маркова в алфавите 𝒜.
Задача 3. Объясните в чём сходство и различие описанной выше машины

Маркова и Рефал-машины. Что в машине Маркова является аналогом:

а). начальных данных?

б). вызова функции?

в). поля зрения?

г). активного вызова функции?

д). аварийной остановки «отождествление невозможно»?

Что в Рефал-машине является аналогом заключительной формулы подстанов-

ки?

Пример: Пусть 𝒜 = {𝑏, 𝑐}. Пусть 𝐺 – свободная алфавитная полугруппа

над 𝒜. Рассмотрим схему: {︃
𝑏 → ∙
𝑐 → 𝑐

Определяемый этой схемой нормальный алгорифм перерабатывает всякое

𝑝0 ∈ 𝐺, содержащее хотя бы одно вхождение буквы 𝑏, в элемент 𝐺, который
получается вычёркиванием в 𝑝0 самого левого вхождения буквы 𝑏. Пустая

последовательность перерабатывается в саму себя. Алгорифм неприменим к

непустым последовательностям, не содержащим буквы 𝑏.

Задача 4. Приведите конструктивное доказательство следующего утвер-

ждения. Если некоторый алгоритм 𝐴 может быть описан схемой Маркова,

то существует программа 𝑈 , написанная на базисном Рефале такая, что для

всех 𝑥 из множества определения функции 𝐹 , соответствующей алгоритму

𝐴, значение 𝐹 (𝑥) может быть вычислено посредством программы 𝑈 , то есть

𝐹 (𝑥) = 𝑈(𝑥). Под конструктивностью мы здесь понимаем явное предъявление

программы 𝑈 .

Задача 5. Реализуйте машину Маркова посредством Рефал-машины.

Как связана данная задача с предыдущей? Проверьте работу построенной вами

программы на всех алгорифмах Маркова, схемы которых рассматриваются в

задачах или схемы которых требуется построить в задачах (данной лекции и

домашнего задания).

6.3. Домашнее задание.

Задача 6. Пусть 𝒜 – произвольный алфавит. Пусть 𝐺 – свободная алфа-

витная полугруппа над 𝒜. Рассмотрим (сокращённо записанную) схему алго-

рифма в алфавите ℬ = 𝒜 ∪ {𝛼, 𝛽}, где 𝛼, 𝛽 /∈ 𝒜.
⎧
⎪⎪⎪⎪⎪⎪⎪⎪⎪⎨
⎪⎪⎪⎪⎪⎪⎪⎪⎪⎩

𝛼 𝛼 → 𝛽

𝛽 𝜉 → 𝜉 𝛽 (𝜉 ∈ 𝐴)

𝛽 𝛼 → 𝛽

𝛽 → ∙
𝛼 𝜂 𝜉 → 𝜉 𝛼 𝜂 (𝜂, 𝜉 ∈ 𝐴)

→ 𝛼

Эта схема определяет некоторый нормальный алгорифм 𝑅𝑒𝑣 в алфавите ℬ.
Докажите, что ∀𝑝 ∈ 𝐺. 𝑅𝑒𝑣(𝑝) = 𝑝. Где последовательность 𝑝 получена из 𝑝
обращением: если 𝑝 = 𝑎1 . . . 𝑎𝑛, то 𝑝 = 𝑎𝑛 . . . 𝑎1.

Задача 7. Пусть алфавит 𝒜 не содержит букву 𝛼, и пусть ℬ = 𝒜 ∪ {𝛼}.
Пусть 𝐺 – свободная алфавитная полугруппа над 𝒜. Построить нормальный

алгорифм Маркова в ℬ, стирающий первую букву во всякой непустой последо-

вательности 𝑝 ∈ 𝐺.

Задача 8. Пусть алфавит 𝒜 не содержит букв 𝛼, 𝛽, 𝛾, и пусть

𝒞 = 𝒜 ∪ {𝛼, 𝛽, 𝛾}. Пусть 𝐺 – свободная алфавитная полугруппу над 𝒜. По-

строить нормальный алгорифм Маркова 𝐹 в 𝒞 такой, что ∀𝑝 ∈ 𝐺 было бы

выполнено равенство 𝐹 (𝑝) = 𝑝 𝑝.

Задача 9. Даны две последовательности (два данных Рефала) 𝑥1, . . . , 𝑥𝑛

и 𝑦1, . . . , 𝑦𝑘. Требуется построить последовательность термов 𝑧1, . . . , 𝑧𝑚, со-
стоящую из совпадающих и имеющих равные порядковые номера термов двух

исходных последовательностей. Порядок термов 𝑧𝑖 друг относительно друга не

изменять.

Задача 10. Длина 𝐿𝑛(d) объектного Рефал-выражения d определяется ин-

дуктивно:

а). длина пустого выражения равна нулю;

б). 𝐿𝑛(t.x e.d) = 1 + 𝐿𝑛(e.d).

Определить на Рефале функцию, значение которой есть длина данного объ-

ектного выражения.

Задача 11. Размер 𝑆𝑖𝑧𝑒(d) объектного Рефал-выражения d определяется

индуктивно:

а). размер пустого выражения равен нулю;

б). 𝑆𝑖𝑧𝑒(s.x e.d) = 1 + 𝑆𝑖𝑧𝑒(e.d);

в). 𝑆𝑖𝑧𝑒((e.x) e.d) = 2 + 𝑆𝑖𝑧𝑒(e.x) + 𝑆𝑖𝑧𝑒(e.d).

Определить на Рефале функцию, значение которой есть размер данного объ-

ектного выражения.

§ 7. Вызов функции по определению

(Лекция №7)

Кроме вызова функции по имени <F e.args>, то есть через указание име-

ни данного определения функции, а не самого определения, синтаксис Рефала

позволяет определить функцию непосредственно в точке её вызова, если обра-

щение к такой функции в программе происходит лишь один раз. Начиная с

этого места нашего изложения, мы выходим за рамки базисного Рефала.

7.1. Блоки. Вызов функции по определению иногда бывает удобен с точки

зрения прозрачности исходного текста программы. Напомним, что в базисном

Рефале определение функции имеет вид последовательности предложений, за-

ключённой в фигурные скобки; а имя функции ставится перед её определением:

FunctionName {

sentence1;

. . . . . . . . .

sentence𝑛;

}

где левая и правая части предложений разделены знаком равенства. Таким

образом, определение само по себе (без имени) есть:

{

sentence1;

. . . . . . . . .
sentence𝑛;

}

Если мы хотим определить функцию непосредственно в точке её вызова, то

вместо имени функции (его у нас теперь нет) мы пишем сам текст определе-

ния, а аргументы этого вызова пишем перед фигурной скобкой, открывающей

данное определение, отделив их от скобки знаком двоеточия «:». Непосред-

ственно за вызовом функции по определению всегда ставится точка с запятой

«;». Например, вызов

’abcd’ (e.x) : {

e.y (e.y) = True;

e.y (e.z) = False;

};

определяет, совпадает ли значение переменной e.x со строкой ’abcd’, а вызов

<F e.w> (e.x) : {

e.y (e.y) = True;

e.y (e.z) = False;

};

определяет, совпадает ли значение результата вызова (по имени) функции F со

значением переменной e.x.

Теперь мы расширим понятие Рефал-предложения: кроме предложений, до-

пускаемых базисным Рефалом, введём предложения, в которых левая часть

отделяется от правой части запятой «,» (а не знаком равенства, как в базис-

ном Рефале). Мы потребуем, чтобы в этом случае в правой части предложе-

ния (после запятой) обязательно стоял один вызов функции по определению, и

только он. Естественно, мы допускаем вызовы по определению внутри любого

определения – поименованного или непоименованного.

Пример №1: Следующая функция S-Type определяет, какого типа символ

подан ей в качестве аргумента, и сообщает об ошибке, если аргумент не явля-

ется Рефал-символом. Мы предварительно, используя встроенную функцию

Lower, преобразуем все прописные буквы имени данного символа в строчные.

S-Type {

s.x, <BelongsTo <Lower s.x> (<Letters>)>: {

True = Letter;

False = Other;

};

e.y = "Wrong argument: " e.y;

}

BelongsTo {

s.x (e.y s.x e.z) = True;

s.x (e.y) = False;

}

Letters {

= ’abcdefghijklmnopqrstuvwxyz’ ’абвгдеёжзийклмнопрстуфхцчшщъыьэюя’;

}

Заметим, что <S-Type ’f’> ̸= <S-Type f>. Причина сего в том, что ’f’ есть

символ-буква, а f есть символ-слово, имя которого состоит из одной буквы.

Рассмотрим общий вид Рефал-предложения с вызовом функции по опреде-

лению:

pattern, argument : {

sentence1;

. . . . . . . . .
sentence𝑛;

};

Множество переменных, входящих в образец pattern, может пересекаться с

множеством переменных входящих в некоторое предложение sentence𝑖. На-

пример,

t.x e.y, <F e.y>: {

t.x e.z = e.z;

e.y e.y e.z = e.z e.z;

e.z = e.y t.x;

};

Здесь первые два предложения (функции, вызванной по определению) содер-

жат в левых частях переменные из образца t.x e.y основного предложения

(вызывающего функцию по определению), а третье предложение содержит в

правой части переменные из того же образца. Как понимать такую ситуацию?

Какой смысл приписать подобному синтаксису? Семантика подобного синтак-

сиса в Рефале следующая: значения всех переменных, которые определились

до запятой «,», Рефал-машина подставляет и в аргумент, и в само определение

вызова функции по определению; и только после этого вычисляет указанный

вызов (естественно, предварительно вычислив его аргументы). Следовательно,

в третьем предложении

e.z = e.y t.x;

нет синтаксической ошибки, ибо в момент вызова оно превратится в

e.z = e.y0 t.x0;

где e.y0 и t.x0 конкретные данные (константы) – значения переменных e.y и

t.x.

Другими словами можно сказать, что определение функции, вызываемой

по определению, может быть параметризовано. В рассмотренном выше при-

мере определение функции зависит от параметров (неизвестных данных) e.y,

t.x. Сущности e.y, t.x являются синтаксическими понятиями – переменными

с точки зрения вызывающей функции, но семантическими понятиями – пара-

метрами, с точки зрения вызываемой функции.

Пример №2: Следующая функция

a-z {

e.1 ’a’ e.2, e.2: {

e.3 ’z’ e.4 = (e.1) ’a’ e.3 ’z’ (e.4);

e.3 = <Prout ’No substring a-z found.’>;

};

e.1 = <Prout ’No \’a\’ found.’>;

}

выделяет первое вхождение в данном тексте строки, начинающейся на ’a’ и

кончающейся на ’z’. В последнем предложении мы «экранировали» две оди-

нарные кавычки символом «\», так как они заключены в другую пару одинар-

ных кавычек.

Задача 1. Рассмотрим другое определение функции из примера №2:

a-z-1 {

e.1 ’a’ e.2 ’z’ e.3 = (e.1) ’a’ e.2 ’z’ (e.3);

e.3 = <Prout ’No substring a-z found.’>;

}

Объясните, почему определение a-z более эффективно, чем определение a-z-1.

То есть найдется Рефал данное d0 такое, что время вычисления вызова

<a-z d0> будет значительно меньше времени вычисления вызова <a-z-1 d0>,

но не существует объектного выражения d такого, что время вычисления вы-

зова <a-z d> будет значительно больше времени вычисления вызова

<a-z-1 d>?

Вызов функции по определению в Рефале принято называть блоком. Присут-

ствие блока в определении функции F приводит к неопределённости понятия

шага Рефал-машины при исполнении F: шаг функции F ещё не завершился,

а уже внутри него начинают вычисляться вызовы функции, готовящие аргу-

менты – входные данные для блока, да и сам блок вычисляется до окончания

рассматриваемого шага функции F.

Задача 2. Провести пошаговый просмотр выполнения программы из при-

мера №2 посредством отладчика. Понять и объяснить, какое действие

Рефал-машины отладчик воспринимает как «шаг» Рефал-5 машины. Изучить

поведение отладчика на последовательности команд p act; com res;.

7.2. Домашнее задание.

Задача 3. Глубину пустого выражения положим равной нулю. Глубина

Depth Рефал-символа также, по определению, равна нулю.

Depth((e.d)) = 1+Depth(e.d)

Глубиной объектного выражения называется максимум глубин всех термов,

входящих в это выражение. Определить на Рефале функцию, значение которой

есть глубина данного объектного выражения.

Задача 4. В данном объектном Рефал-выражении (со скобками) уничто-

жить первое вхождение символа ’a’, если таковое существует.

Задача 5. Даны две конечные неубывающие последовательности макро-

цифр. Построить неубывающую последовательность, элементами которой яв-

ляются все элементы (включая их кратные вхождения) двух данных последо-

вательностей.

Задача 6. Нарисовать ёлочку на экране компьютера средствами псевдо-

графики (посредством символов клавиатуры). Высота ёлочки должна быть

равна высоте экрана.

Задача 7. Дана конечная последовательность макроцифр 𝑥1 = 1, 𝑥𝑛+1 =
1 + 𝑥𝑛. Построить все различные перестановки данной последовательности.

Задача 8. Дана симметрическая группа перестановок 𝑆𝑛. Придумать

представление элементов этой группы во множестве данных Рефала и реали-

зовать групповые операции умножения и построения элемента, обратного к

данному элементу.

Задача 9. Рассмотрим полугруппу по умножению квадратных матриц раз-

мера 𝑛 × 𝑛, состоящих из целых чисел. Придумать представление элементов

этой полугруппы как данных Рефала и реализовать операцию умножения двух

матриц.

§ 8. Рефал как метаязык программирования

(Лекция №8)

Язык Рефал ориентирован на преобразование текстов. Рефал изначаль-

но разрабатывался В.Ф. Турчиным как метаязык, – то есть инструмент для

анализа и преобразования текстов, написанных на других языках, в частно-

сти, на языках программирования. Программы (тексты на языках програм-

мирования) обычно анализируются либо с целью их перевода на другие языки

программирования, либо с целью их исполнения. В обоих случаях семантика

соответствующего языка программирования должна быть определена в терми-

нах Рефала.

Объектами преобразований Рефал-программ могут быть программы, напи-

санные также на языке Рефал. В частности, из алгоритмической полноты

Рефала следует, что функция семантики Рефала может быть описана в терми-

нах самого Рефала. Рассмотрим некоторые проблемы, связанные с понятием

метаязыка.

Перед прочим заметим, что примером метаязыка является русский язык, а

примером анализа текстов на русском языке посредством самого языка явля-

ется предложение, которое вы сейчас читаете.

При самоанализе возникает проблема с синтаксисом: если мы пишем текст

о синтаксисе русского языка, то, например, знаки препинания мы вынуждены

обозначать не самими этими знаками как таковыми, а словами, их обозначаю-

щими – иначе мы нарушим правила самого синтаксиса, о котором рассуждаем.

И вместо предложения: «В конце предложения нужно ставить точку.» у нас

получится предложение: «В конце предложения нужно ставить ..», которое

заканчивается знаком горизонтального двоеточия, отсутствующим в русском

языке.

В Рефале обсуждаемая проблема проявляется, например, в том, что если

требуется, при анализе одной Рефал-программы посредством другой Рефал-

программы, определить имя некоторой переменной s.x, то переменную s.x (а

не её значение) невозможно передать напрямую (как значение аргумента) для

анализа:

$ENTRY Go {

= <Analyze s.x>;

}

Снова возникает синтаксическая ошибка: переменная в правой части предло-

жения, которой нет в левой части этого предложения. Как и в русском языке,

в подобных случаях мы вынуждены представлять одни понятия языка посред-

ством других. Например, так:

$ENTRY Go {

= <Analyze (Variable ’s’ x)>;

}

Представление одних понятий языка посредством других называется кодиров-

кой.

Кодировка должна позволять восстанавливать смысл закодированного по-

нятия, как и позволять узнавать, является ли то или иное понятие кодировкой

(или частью кодировки) другого понятия, или оно представляет само себя. В

противном случае неизбежно возникнет путаница. В предложении «Рассмот-

рим точку 𝐴.» слово точка представляет само себя, а не знак, которым за-

канчивается повествовательное предложение. То же самое можно сказать и

о слове точка как в предыдущем предложении (второе вхождение), так и в

данном предложении. В нашем примере мы смогли определить смысл слова

«точка» из контекста.

Язык программирования 𝐿 = {𝐷,𝑃, 𝑆}, где 𝐷 – множество данных, 𝑃 – мно-

жество программ, 𝑆 – функция семантики, назовём метаязыком программи-

рования, если в нём фиксирована инъективная функция кодировки 𝜈 : 𝑃 → 𝐷.

Инъективность здесь нужна для однозначного восстановления смысла по-

нятий. Далее, для лучшей читаемости, мы будем обозначать отображение ко-

дировки подчёркиванием. Например, 𝜈(s.x) = s.x .

Задача 1. Дана матрица 𝑀 размера 𝑛 × 𝑛, состоящая из нулей и единиц.

Придумать представление𝑀 в множестве данных Рефала и написать програм-

му, результатом которой является строка длины 𝑛, состоящая из нулей и единиц
и не являющаяся строкой матрицы 𝑀 .

8.1. Самоприменимые программы. Программу p назовём самоприме-

нимой, если вызов <p p> не циклится (работает конечное время).

Пусть p ∈ 𝑃 . Рассмотрим функцию 𝜑 : 𝐼𝑚(𝜈) → {finite, infinite}2

𝜙(𝑝) =

{︃
finite, если <p p> заканчивает работу в конечное время;

infinite, если вызов <p p> циклится.

Везде ранее мы рассматривали только вычислимые (конструктивные)

функции. Другими словами, функции которые можно определить на алго-

ритмически полном языке. Покажем, что функция 𝜙 не является вычислимой.

Теорема: Не существует программы, которая распознавала бы самопри-

менимые программы.

Доказательство будем вести методом от обратного. Предположим, что су-

ществует программа q, которая реализует функцию 𝜙. Тогда рассмотрим про-

грамму f такую, что

<f p> = finite, если <q p> = infinite,

и <f p> циклится, если <q p> = finite.

Рассмотрим вызов <f f>. Предположим что <f f> = finite, тогда по опре-

делению f <q f> = infinite и, следовательно, теперь уже по определению q,

программа f не является самоприменимой, что противоречит нашему предпо-

ложению.

Следовательно, вызов <f f> циклится. Но тогда по определению f,

<q f> = finite и, следовательно, по определению q, программа f является

самоприменимой, что противоречит тому, что вызов <f f> циклится.

Таким образом, оба из логически возможных поведений исполнения вызо-

ва <f f> приводят к противоречию. И наше предположение о существовании

программы q, реализующей функцию 𝜙, неверно. Доказательство теоремы за-

кончено.

Задача 2. Объясните, каким образом приведённое выше доказательство

связано с задачей №1 про матрицу.

Задача 3. Докажите, что нельзя написать программу, которая для любой

программы p определяла бы, что существует 𝑑0 ∈ 𝐷 такое, что <p 𝑑0> циклит-

ся.
2𝐼𝑚(𝜈) – множество значений (образ) отображения 𝜈.

Множество 𝑀 называется счётным, если его можно пересчитать. Другими

словами: существует биекция N → 𝑀 .

Задача 4. Докажите, что множество всех действительных чисел из интер-

вала (0, 1) несчётно. Объясните, каким образом утверждение данной задачи

связано утверждением о несуществовании программы, которая распознавала

бы самоприменимые программы.

8.2. Функция применения. Приведём пример встроенной

мета-функции Рефала-5. Вызовы

<Mu s.fname e.args>

и

<Mu (e.fname) e.args>

вычисляют значение функции с именем, переданным через переменную, на

аргументе e.args. В первом случае имя вычисляемой функции есть символ

Рефала, во втором случае имя функции представлено последовательностью

букв.

Задача 5. Даны имя функции и последовательность объектных термов.

Требуется определить функцию, значение которой есть последовательность

значений данной функции на каждом из термов данной последовательности.

Порядок значений должен совпадать с порядком аргументов.

Задача 6. Даны имя функции и последовательность корневых деревьев

(объектных Рефал-выражений). Требуется определить программу, заменяю-

щую каждый лист данных деревьев значением данной функции на этом листе.

8.3. Домашнее задание.

Задача 7. Докажите, что ниже перечисленные множества счётны, опреде-

лите соответствующие биекции на Рефале:

а). множество целых чисел Z;
б). множество Z2;

в). множество Z3;

г). множество Z𝑘0 , где 𝑘0 фиксированное натуральное число.

Задача 8. Является ли счётным множество данных Рефала? Если ответ

положителен, то определите соответствующую биекцию на Рефале.

Задача 9. Докажите, что нельзя написать программу, которая для любой

программы p определяла бы, что для всех 𝑑 ∈ 𝐷. <p 𝑑> = 𝑑.

Задача 10. Дан объектный терм 𝑇0 и последовательность имён функций.

Требуется определить функцию, значение которой есть значение композиции

данных функций, первая из которых вычисляется на 𝑇0.

§ 9. Рекурсивные условия

(Лекция №9)

Ранее мы ввели в синтаксис Рефал-программ новое понятие – запятую, тем

самым выйдя за рамки базисного Рефала. Кроме отделения левой части от

правой части предложения, при вызове функции по определению знак запя-

той используется для организации рекурсивного условия выбора предложения

внутри одной функции. В обоих случаях знак запятой можно понимать как

логическую связку – конъюнкцию (то есть «и»).

9.1. Другая интерпретация знака запятой. Вернёмся к нашему приме-

ру выделения в данной строке подстроки, начинающейся на ’a’ и кончающейся

на ’z’. Ранее мы запрограммировали его, используя блок (вызов функции по

определению).

a-z {

e.1 ’a’ e.2, e.2: {

e.3 ’z’ e.4 = (e.1) ’a’ e.3 ’z’ (e.4);

e.3 = ’No substring a-z found.’;

};

e.1 = ’No \’a\’ found.’;

}

Смысл пары запятая-двоеточие в первом предложении

e.1 ’a’ e.2, e.2:

можно понимать и так: если аргумент вызова функции a-z является Рефал-

выражением вида e.1 ’a’ e.2 и значение переменной e.2 такое, что . . . (да-
лее см. определение блока). Или ещё так: если аргумент вызова функции a-z

является объектным Рефал-выражением вида e.1 ’a’ e.2, где значение пере-

менной e.2 такое, что . . . (далее см. определение блока). Именно такой смысл

пары запятая-двоеточие вкладывается в «, expression :» при введении в рас-

ширенный Рефал-рекурсивных условий.

9.2. Рекурсивные условия выбора предложения. В базисном Рефа-

ле Рефал-машина использует структуру образца и алгоритм сопоставления

данных-аргументов функции F с образцом для выбора предложения из после-

довательности предложений, определяющей тело функции F. Механизм отож-

дествления не является «алгоритмически полным». (Например, он не может

иметь бесконечных циклов.) Используем отмеченный выше смысл синтакси-

ческой конструкции «, expression :» для введения алгоритмически полного

инструмента выбора Рефал-предложения.

В расширенном Рефале Рефал-предложение может иметь один из следую-

щих видов:

1. pattern = expression;

2. pattern , arguments : { block };

3. pattern conditions right_part;

где right_part есть либо «= expression», либо «, arguments : { block }»,

а

conditions ::= , expression : pattern conditions | empty

Причём expression имеет тот же вид что и правая часть предложения в

базисном Рефале, и удовлетворяет тем же синтаксическим ограничениям: в

expression допускаются только переменные, которые уже встречались в рас-

сматриваемом предложении левее данного expression.



158 А.П. НЕМЫТЫХ

Мы начнём с простейшего примера, показывающего, что даже при отсут-

ствии рекурсии в выборе предложения, новая конструкция условия позволяет

писать программы более компактно.

Пример №1: Пусть функция Look-For-I(𝑥1, 𝑥2, 𝑥3), где 𝑥𝑖 – строки симво-

лов, есть предикат, отвечающий на вопрос «Принадлежит ли символ ’I’ хотя

бы одной строке 𝑥𝑖?». Тогда в базисном Рефале определение Look-For-I, если

не использовать вспомогательных функций, выглядит так:

Look-For-I {

(e.x ’I’ e.x1) (e.x2) (e.x3) = True;

(e.x1) (e.y ’I’ e.x2) (e.x3) = True;

(e.x1) (e.x2) (e.z ’I’ e.x3) = True;

(e.x1) (e.x2) (e.x3) = False;

}

Используя блок, его можно упростить следующим образом:

Look-For-I {

(e.x1) (e.x2) (e.x3), e.x1 e.x2 e.x3 : {

e.x ’I’ e.y = True;

e.z = False;

};

}

Использование условия позволяет написать эту программу ещё короче:

Look-For-I {

(e.x1) (e.x2) (e.x3), e.x1 e.x2 e.x3 : e.x ’I’ e.y = True;

e.z = False;

}

9.2.1. Откаты при выборе предложения.

Пример №2: Дана строка e.string и лес деревьев e.forest. Требу-

ется найти самое левое дерево, среди листов которого содержится самый

левый символ строки e.string, из тех которые являются листьями данного

дерева. Определим соответствующую программу, используя рекурсию в

выборе предложений.

Tree {

e.x s.leaf e.string (e.y t.tree e.forest)

, <Get-leaves t.tree>: e.z s.leaf e.leaves = t.tree;

e.x (e.forest) = Fail;

}

Get-leaves {

e.x (e.y) e.z = e.x <Get-leaves e.y e.z>;

e.x = e.x;

}

Здесь рекурсивно происходят откаты при выборе Рефал-машиной предложения

функции Tree. Два образца

e.x s.leaf e.string (e.y t.tree e.forest) и s.z s.leaf e.leaves

имеют общую переменную s.leaf и взаимодействуют при работе алгоритмов

отождествления. Согласно общему правилу Рефала при решении уравнения

e.x s.leaf e.string (e.y t.tree e.forest) = string0 (forest0)

, где string0 и forest0 – данные Рефала, из всего множества решений 𝑀𝑖

1. выбираются те решения, в которых e.x приняла значение с наименьшей

возможной длиной (количеством деревьев в e.x0),

2. если такой выбор не приводит к однозначности решения, тогда из вы-

бранного подмножества решений выбираются те решения, в которых e.y

приняла значение с наименьшей возможной длиной (количеством дере-

вьев в e.y0). Последний выбор всегда приведёт к однозначности, если

множество решений не пусто.

Таким образом, в результате решения указанного выше уравнения Рефал-

машина имеет значения переменных t.tree = t.tree0, s.leaf = s.leaf0 и

может вычислить вызов <Get-leaves t.tree0>. Пусть результат этого вычис-

ления равен leaves0, тогда Рефал-машина решает уравнение, определённое

условием в первом предложении функции Tree:

e.z s.leaf e.leaves = leaves0
если это уравнение имеет решение, тогда Рефал-машина нашла дерево t.tree0
с требуемым свойством. Иначе Рефал-машина возвращается к уравнению

e.x s.leaf e.string (e.y t.tree e.forest) = string0 (forest0)

и считает, что выбранное ранее его решение

e.x0, s.leaf0, e.string0, e.y0, t.tree0, e.forest0
не удовлетворяет условию выбора данного предложения. Пусть 𝑙𝑛(e.y0) есть

длина (количество термов-деревьев в) e.y0, тогда множество решений данного

уравнения сужается условием 𝑙𝑛(e.y) > 𝑙𝑛(e.y0).

∙ Если получившееся подмножество не пусто, то обозначим его 𝐾𝑖+1 ⊂ 𝑀𝑖

(𝐾𝑖+1 ̸= 𝑀𝑖), переименуем его 𝑀𝑖 := 𝐾𝑖+1 и переходим к пункту 1).

∙ Иначе Рефал-машина снова возвращается к решению уравнения

e.x s.leaf e.string (e.y t.tree e.forest) = string0 (forest0)

Множество решений 𝑀𝑖 данного уравнения сужается условием

𝑙𝑛(e.x) > 𝑙𝑛(e.x0). Обозначим получившееся подмножество 𝑄𝑖+1 ⊂ 𝑀𝑖

(𝑄𝑖+1 ̸= 𝑀𝑖), переименуем его 𝑀𝑖 := 𝑄𝑖+1 и переходим к пункту 1).

Задача 1. Доказать, что определённая выше функция Tree заканчивает

работу за конечное время при любых значениях её аргументов.

Задача 2. Каковы будут значения переменных e.x и e.y при первом вызове

функции Get-leaves? Зависят ли эти значения от аргументов вызова функции

Tree?

Задача 3. Чему равно максимальное число откатов, которое сделает

Рефал-машина, при поиске нужного решения уравнения

e.x s.leaf e.string (e.y t.tree e.forest) = string0 (forest0) ?

Вставим в нашу программу отладочную печать, чтобы проследить работу

Рефал-машины в деталях.

$ENTRY Go {

= <Prout ’\nThe result: ’ <Tree A B C ((E) (D (B)) ((B) (N C)))>>;

}

Tree {

e.x s.leaf e.string (e.y t.tree e.forest),

<Prout ’Try the leaf: ’ s.leaf ’ in the tree: ’ t.tree

’ \n\t , where the length of e.y=’ e.y ’ is ’ <Ln e.y>>

<Prout ’ \t and the length of e.x=’ e.x ’ is ’ <Ln e.x>>

<Get-leaves t.tree>: e.z s.leaf e.leaves = t.tree;

e.x (e.forest) = Fail;

}

Get-leaves {

e.x (e.y) e.z = e.x <Get-leaves e.y e.z>;

e.x = e.x;

}

Ln {

= 0;

t.term e.x = <+ (<Ln e.x>) 1>;

}

В программе мы использовали символ перевода строки ’\n’ и символ го-

ризонтальной табуляции ’\t’, также мы воспользовались тем, что значение

вызова <Prout e.args> всегда равно пустому выражению. В результате ис-

полнения этой программы на экране компьютера имеем:

Try the leaf: A in the tree: (E )

, where the length of e.y= is 0

and the length of e.x= is 0

Try the leaf: A in the tree: (D (B ))

, where the length of e.y=(E ) is 1

and the length of e.x= is 0

Try the leaf: A in the tree: ((B )(N C ))

, where the length of e.y=(E )(D (B )) is 2

and the length of e.x= is 0

Try the leaf: B in the tree: (E )

, where the length of e.y= is 0

and the length of e.x=A is 1

Try the leaf: B in the tree: (D (B ))

, where the length of e.y=(E ) is 1

and the length of e.x=A is 1

The result: (D (B ))

Задача 4. Выполните данную выше программу с отладочной печатью на

вашем компьютере. Вставьте дополнительную печать, показывающую как из-



ЛЕКЦИИ ПО ЯЗЫКУ ПРОГРАММИРОВАНИЯ РЕФАЛ 161

меняется при выборе предложения длина значения переменной e.z. Объясните

полученный вами результат.

9.2.2. Локальное присваивание. Синтаксическую конструкцию условия

можно также использовать для устранения повторных вычислений. В этом

случае само условие тривиально – всегда истинно, и мы используем его кон-

струкцию как конструкцию локального присваивания.

Пример №3: Пусть нам нужно определить функцию на множестве целых

чисел g(𝑛) = (𝑛− 1) * (𝑛− 1). Определим g в базисном Рефале:

g {

e.n = <* (<- (e.n) 1>) <- (e.n) 1>>;

}

Использование синтаксической конструкции условия позволяет устранить

повторное вычитание 1.

g {

e.n, <- (e.n) 1>: e.m = <* (e.m) e.m>;

}

Задача 5. Переопределить данную в примере №3 функцию g в базисном

Рефале, но без повторного вычитания 1.

Присутствие условия в определении функции F может привести к неопреде-

лённости понятия шага Рефал-машины при исполнении F (как и присутствие

блока в F): шаг функции F ещё не завершился, а уже внутри него начинают

вычисляться вызовы функции, готовящие аргумент для этого условия.

Задача 6. Провести пошаговый просмотр выполнения программы из при-

мера №2 посредством отладчика. Понять и объяснить какое действие Рефал-

машины отладчик воспринимает как «шаг» Рефал-5 машины. Изучить пове-

дение отладчика на последовательность команд p act; com res.

9.3. Домашнее задание.

Задача 7. Переопределить данную в примере №2 функцию Tree, ограни-

чившись синтаксисом базисного Рефала.

Задача 8. Дано объектное Рефал-выражение 𝑑0, в которое могут входить

структурные скобки. Определить функцию 𝑓 поиска первого вхождения знака

аддитивной операции (’+’ или ’-’) на верхнем уровне скобочной структуры

данного 𝑑0. Значением 𝑓 должно быть исходное 𝑑0, в котором часть предше-

ствующая найденному знаку аддитивной операции заключена в структурные

скобки. Если требуемого знака не существует, то функция 𝑓 должна выдавать

не изменённое 𝑑0.

Задача 9. Дан лес деревьев (объектное Рефал-выражение со скобками).

Требуется определить на Рефале функцию, переставляющую листья деревьев

в обратном порядке, если листья находятся на нечётной глубине, и не меняю-

щую порядок листьев, находящихся на чётной глубине. Структуру деревьев не

изменять. Задачу решить в двух вариантах:

а). Локально, то есть переставляются в каждом дереве листья, находящиеся

на ветках одного и того же ближайшего к ним узла-ветвления. Напри-

мер,

(a b c) d t (f g 1 (3 4) 5) ⇒ (c b a) d t (5 1 g (3 4) f);

б). Глобально, то есть последовательности листьев, находящихся на глубине

2 * 𝑛 + 1, приписываются друг к другу согласно следованию – слева на-

право. После этого элементы полученной последовательности перестав-

ляются в обратном порядке и затем развешиваются на соответствующие

ветви деревьев.

Например, для леса (a b c) d t (f g 1 (3 4) 5) имеем последова-

тельность листьев, соответствующую глубине 1: a b c f g 1 5. После

перестановки элементов этой последовательности она преобразуется в

5 1 g f c b a. Теперь развешиваем листья на деревья и получаем:

(5 1 g) d t (f c b (3 4) a).

§ 10. Сложность Колмогорова

(Лекция №10)

Рефал ориентирован на преобразование текстов. Какие тексты являются

простыми, а какие сложными? Классическое рифмованное стихотворение вы-

учить наизусть много проще, чем белый стих. Кроме того, понятие сложности

текста T зависит от нашего понимания T. Запоминание доказательства теоре-

мы при подготовке к экзамену является делом безнадёжным, когда студент

не понимает этого доказательства; в то же время, если доказательство поня-

то, то оно легко восстанавливается, исходя из логики взаимозависимости его

шагов. Великий русский математик Андрей Николаевич Колмогоров предло-

жил аппроксимировать понятие взаимозависимости частей данного текста

посредством рекурсии.

10.1. Сложность текста. Иногда возможно длинный текст описать по-

средством короткого. Например, текст

миллион слов ’рефал’,

состоящий из 20 символов, является описанием текста длиной пять миллио-

нов символов. Ранее мы уже пробовали описывать русскую народную сказку

«Репка» посредством как можно более короткой Рефал-программы.

Задача 1. Дано стихотворение Владимира Солоухина, рассмотренное на-

ми на лекции №5 – задача №9. Обозначим его 𝑇1. Создать текст 𝑇2, состоящий

из одних нулей, длина которого равна длине текста 𝑇1. Из текста данной лек-

ции №5 вырезать кусок 𝑇3; длина текста 𝑇3 должна равняться длине текста

𝑇1. Сжать тексты 𝑇1, 𝑇2, 𝑇3 всеми известными вам упаковщиками («архива-

торами»), составить таблицу (одна строчка для каждого упаковщика) длин

получившихся текстов и сравнить эти длины.

Пусть p – какая-нибудь программа. Сложностью текста 𝑇 при способе

описания p называется наименьшая длина текста 𝑍 такого, что <p 𝑍> = 𝑇 .
При этом текст 𝑍 называется описанием текста 𝑇 .

Иными словами, мы рассматриваем все сжатые версии текста 𝑇 , из которых

данный распаковщик p может восстановить 𝑇 , и среди них выбираем самое

короткое описание. Например,

миллион слов ’рефал’

и

пятьсот тысяч слов ’рефалрефал’

– два описания одного текста, но первое описание короче.

Некоторые упаковщики создают самораспаковывающиеся архивы. В таких

архивах содержится и распаковщик, и заархивированный текст. Текст распа-

ковщика должен быть каким-то образом отделён от заархивированного текста.

10.2. Структурные скобки и кодирование разделения аргументов.

С подобной проблемой мы уже встречались: в Рефале все функции одноместны

(имеют только один аргумент). Мы использовали конструктор структурных

скобок, чтобы имитировать функции нескольких переменных. То есть, мы

кодировали понятие разделения аргументов посредством структурных

скобок. Другими словами, мы структурировали данные Рефала. И, как мы

знаем, структурные скобки позволяют удобно структурировать Рефал-данные

в виде леса корневых деревьев; в этом основная роль структурных скобок.

За такую возможность мы заплатили ограничением алфавита: скобка ( не

может представлять сама себя в тексте Рефал-программы, – синтаксис требует

закодировать её обрамлением в одинарные или двойные кавычки (так ’(’ или

так "(").

Оказывается, можно закодировать понятие разделения аргументов, не вводя

никаких дополнительных ограничений на алфавит, если алфавит содержит по

крайней мере две буквы. Разделить распаковщик и заархивированный текст

можно, например, таким образом:

рраассппааккууюющщааяя ппррооггррааммммааабзапакованный текст

где мы каждую букву исходного текста распаковщика продублировали (после

яя стоят два пробела), а сразу после закодированного текста распаковщика мы

написали буквы аб ; сам запакованный текст никак не изменили.

Задача 2. Докажите, что указанное кодирующее отображение инъективно.

Задача 3. Решите следующую задачу, не используя структурные скобки.

Даны текст и строка. Определить на Рефале функцию, значение которой есть

отрезок данного текста, непосредственно следующий за первым вхождением

данной строки, если такое вхождение существует, и сама данная строка, если

она не входит в текст.

10.3. Теорема Колмогорова. Пусть дан алгоритмически полный язык

программирования 𝐿 (например, Рефал без структурных скобок), на котором

мы намерены реализовывать упаковывающие программы. Пусть 𝐷 есть мно-

жество данных языка 𝐿, а 𝑃 – множество программ языка 𝐿.

Теорема 1. Существует способ, который даёт почти самые короткие опи-

сания всех текстов. То есть, существует программа p такая, что для любого

текста 𝑇 сложность описания 𝑇 посредством p меньше, чем сложность опи-

сания 𝑇 посредством любой другой программы q плюс некоторая константа



164 А.П. НЕМЫТЫХ

Const, зависящая только от q. То есть, Const не зависит от 𝑇 . Обозначим

через 𝜙p(𝑇 ) сложность описания 𝑇 посредством p, тогда

∃𝑝 ∈ 𝑃 ∀𝑞 ∈ 𝑃 ∃Const ∈ R ∀𝑇 ∈ 𝐷. 𝜙p(𝑇 ) < 𝜙q(𝑇 ) + Const

Теорема говорит о том, что можно написать универсальный почти оптималь-

ный упаковщик p. Любой другой упаковщик q будет сжимать исходные тексты

хуже, чем p – с точностью до некоторой константы.

Докажем теорему. Данный язык 𝐿, по условию, алгоритмически полон.

Следовательно, мы можем реализовать на языке 𝐿 универсальную программу

Int ∈ 𝑃 , которая умеет вычислять значение любой другой данной программы

q ∈ 𝑃 (написанной на языке 𝐿) на любом конкретном d0 ∈ 𝐷. Программа q есть

отображение из 𝐷 в 𝐷⊥ (смысл обозначения см. в лекции №2). Следовательно,

Int : 𝑃 ×𝐷 → 𝐷⊥

Условие универсальности программы Int формально выглядит так:

<Int qd> = <q d>

где мы намеренно не разделили программу q и её данные d, – разделение делает

кодировка, обозначенная нами подчёркиванием. Универсальные программы

программисты называют интерпретаторами (interpreter – исполнитель).

Пусть у нас имеется произвольный способ описания q ∈ 𝑃 , тогда длину q𝑍

можно ограничить числом 2 * 𝑘 + 𝑛 + 2, где 𝑘 – длина q, 𝑛 – длина текста 𝑍,
являющего описанием текста 𝑇 при способе q. Далее,

∀𝑍 <Int q𝑍> = <q 𝑍> = 𝑇

то есть пара (Int, q𝑍) распаковывает текст 𝑇 при способе Int. Следователь-

но, по определению описания,

𝜙Int(𝑇 ) < 2 * 𝑘 + 𝑛 + 3

По определению, 𝑍 есть описание 𝑇 при способе q. То есть 𝑛 = 𝜙q(𝑇 ). Следо-
вательно,

∀𝑞 𝜙Int(𝑇 ) < 𝜙q(𝑇 ) + 2 * 𝑘 + 3

где 𝑘 – длина самого способа q (длина исходного текста программы q), которая

зависит только от q и не зависит от 𝑇 . Что и требовалось доказать.

10.4. Домашнее задание.

Задача 4. Выше мы описали кодировку понятия разделения аргументов.

а). Реализуйте описанную кодировку на Рефале.

б). Реализуйте описанную кодировку на Рефале, не используя структурные

скобки.

Задача 5. Ограничим алфавит Рефал-символов тремя символами ’a’, ’b’,

’_’. Получившийся язык обозначим Рефалab_. Натуральные числа в языке

Рефалab_ будем представлять в троичной системе счисления, цифрами в ко-

торой являются символы ’_’, ’a’, ’b’ (’_’ есть 0, ’a’ есть 1, ’b’ есть 2).

Рассмотрим следующую кодировку понятия разделения аргументов. Пусть да-

ны два текста-аргумента X и Y; вычислим длину 𝑘 первого аргумента X. Пусть

𝑁(𝑘) – представление числа 𝑘 в троичной системе счисления с цифрами ’_’,

’a’, ’b’. Рассмотрим пару 𝑁(𝑘) и XY, где мы текст Y приписали к тексту X без

какого-либо разделителя между ними. Закодируем пару 𝑁(𝑘) и XY посредством
удвоения букв в тексте 𝑁(𝑘) и отделением первого элемента пары от второго

двумя неравными буквами (как мы описали выше в нашей лекции).

Докажите, что построенное отображение, кодирующее два текста–аргумента

X и Y, инъективно.

а). Реализуйте описанную кодировку на Рефале.

б). Реализуйте описанную кодировку на Рефале, не используя структурные

скобки.

в). Реализуйте описанную кодировку на языке Рефалab_.

г). Реализуйте описанную кодировку на языке Рефалab_, не используя струк-

турные скобки.

